Chapter 6

Compilation
The model of evaluation introduced in Section 2.3 and formalized in Section 3.6
builds only on the expressions of the Mini-ML language itself. This leads very
naturally to an interpreter in Elf which is given in Section 4.3. Our specification of
the operational semantics is in the style of natural semantics which very often lends
itself to direct, though inefficient, execution. The inefficiency of the interpreter
in 4.3 is more than just a practical issue, since it is clearly the wrong model if we
would like to reason about the complexity of functions defined in Mini-ML. One can
refine the evaluation model in two ways: one is to consider more efficient interpreters
(see Exercises 2.12 and 4.2), another is to consider compilation. In this chapter we
pursue the latter possibility and describe and prove the correctness of a compiler
for Mini-ML.
In order to define a compiler we need a target language for compilation, that is,
the language into which programs in the source language are translated. This target
language has its own operational semantics, and we must show the correctness of
compilation with respect to these two languages and their semantics. The ultimate
target language for compilation is determined by the architecture and instruction
set of the machine the programs are to be run on. In order to insulate compilers
from the details of particular machine architectures it is advisable to design an
intermediate language and execution model which is influenced by a set of target
architectures and by constructs of the source language. We refer to this intermediate level as an abstract machine. Abstract machine code can then itself either be
interpreted or compiled further to actual machine code. In this chapter we take a
stepwise approach to compilation, using two intermediate forms between Mini-ML
and a variant of the SECD machine [Lan64] which is also related to the Categorical
Abstract Machine (CAM) [CCM87]. This decomposition simplifies the correctness
proofs and localizes ideas which are necessary to understand the compiler in its
totality.

The material presented in this chapter follows work by Hannan [HM90, Han91],
both in general approach and in many details. An extended abstract that also
addresses correctness issues and methods of formalization can be found in [HP92].
A different approach to compilation using continuations may be found in Section ??.

6.1

An Environment Model for Evaluation

The evaluation judgment e ,→ v requires that all information about the state of the
computation is contained in the Mini-ML expression e. The application of a function
formed by λ-abstraction, lam x. e, to an argument v thus requires the substitution
of v for x in e and evaluation of the result. In order to avoid this substitution
it may seem reasonable to formulate evaluation as a hypothetical judgment (e is
evaluated under the hypothesis that x evaluates to v) but this attempt fails (see
Exercise 6.1). Instead, we allow free variables in expressions which are given values
in an environment, which is explicitly represented as part of a revised evaluation
judgment. Variables are evaluated by looking up their value in the environment;
previously we always eliminated them by substitution, so no separate rule was
required. However, this leads to a problem with the scope of variables. Consider
the expression lam y. x in an environment that binds x to z. According to our
natural semantics the value of this expression should be lam y. z, but this requires
the substitution of z for x. Simply returning lam y. x is incorrect if this value
may later be interpreted in an environment in which x is not bound, or bound
to a different value. The practical solution is to return a closure consisting of an
environment η and an expression lam y. e. η must contain at least all the variables
free in lam y. e. We ignore certain questions of efficiency in our presentation and
simply pair up the complete current environment with the expression to form the
closure.
This approach leads to the question how to represent environments and closures.
A simple solution is to represent an environment as a list of values and a variable
as a pointer into this list. It was de Bruijn’s idea [dB72] to implement such pointers
as natural numbers where n refers to the nth element of the environment list. This
works smoothly if we also represent bound variables in this fashion: an occurence
of a bound variable points backwards to the place where it is bound. This pointer
takes the form of a positive integer, where 1 refers to the innermost binder and 1
is added for every binding encountered when going upward through the expression.
For example
lam x. lam y. x (lam z. y z)
would be written as
Λ (Λ (2 (Λ (2 1))))
where Λ binds an (unnamed) variable. In this form expressions that differ only
in the names of their bound variables are syntactically identical. If we restrict

attention to pure λ-terms for the moment, this leads to the definition
de Bruijn Expressions
de Bruijn Indices

::= n | ΛD | D1 D2
::= 1 | 2 | . . .

D
n

Instead of using integers and general arithmetic operations on them, we use only
the integer 1 to refer to the innermost element of the environment and the operator
↑ (read: shift, written in post-fix notation) to increment variable references. That
is, the integer n + 1 is represented as
1 ↑···↑ .
| {z }
n times

But ↑ can also be applied to other expressions, in effect raising each integer in the
expression by 1. For example, the expression
lam x. lam y. x x
can be represented by
Λ (Λ ((1↑) (1↑)))
or
Λ (Λ ((1 1)↑)).
This is a very simple form of a λ-calculus with explicit substitutions where ↑ is the
only available substitution (see [ACCL91]).
Modified de Bruijn Expressions

F

::= 1 | F ↑ | ΛF | F1 F2

We use the convention that the postfix operator ↑ binds stronger than application
which in turn binds stronger that the prefix operator Λ. Thus the two examples
above can be written as Λ Λ 1↑ 1↑ and Λ Λ (1 1)↑, respectively.
The next step is to introduce environments. These depend on values and vice
versa, since a closure is a pair of an environment and an expression, and an environment is a list of values. This can be carried to the extreme: in the Categorical
Abstract Machine (CAM), for example, environments are built as iterated pairs and
are thus values. Our representation will not make this identification. Since we have
simplified our language to a pure λ-calculus, the only kind of value which can arise
is a closure.
Environments
η ::= · | η, W
Values W ::= {η; F }
We write w for parameters ranging over values. During the course of evaluation, only
closures over Λ-expressions will arise, that is, all closures have the form {η; ΛF 0}
(see Exercise 6.2).
The specification of modified de Bruijn expressions, values, and environments
is straightforward. The abstract syntax is now first-order, since the language does
not contain any name binding constructs.

exp’

: type.

1
^
lam’
app’

:
:
:
:

%name exp’ F f.

exp’.
exp’ -> exp’. %postfix 20 ^.
exp’ -> exp’.
exp’ -> exp’ -> exp’.

% Environments and values
env
val

: type.
: type.

%name env N.
%name val W w.

empty
,

: env.
: env -> val -> env.

clo

: env -> exp’ -> val.

%infix left 10 ,.

There are two main judgments that achieve compilation: one relates a de Bruijn
expression F in an environment η to an ordinary expression e, another relates a
value W to an expression v. We also need an evaluation judgment relating de
Bruijn expressions and values in a given environment.
η`F ↔e
W ⇔v
η ` F ,→ W

F translates to e in environment η
W translates to v
F evaluates to W in environment η

When we evaluate a given expression e using these judgments, we translate it to a de
Bruijn expression F in the empty environment, evaluate F in the empty environment
to obtain a value W , and then translate W to an expression v in the original
language. This is depicted in the following diagram.
e
C :: · ` F ↔ e

?

F

D :: e ,→ v

-v
6
-W

U :: W ⇔ v

D0 :: · ` F ,→ W

The correctness of this phase of compilation can then be decomposed into two
statements. For completeness, we assume that D and therefore e and v are given,
and we would like to show that there exist C, D0 , and U completing the diagram.
This means that for every evaluation of e to a value v, this value could also have
been produced by evaluating the compiled expression and translating the resulting
value back to the original language. The dual of this is soundness: we assume that

C, D0 and U are given and we have to show that an evaluation D exists. That
is, every value which can be produced by compilation and evaluation of compiled
expressions can also be produced by direct evaluation.
We will continue to restrict ourselves to expressions built up only from abstraction and application. When we generalize this later only the case of fixpoint
expressions will introduce an essential complication. First we define evaluation of
de Bruijn expressions in an environment η, written as η ` F ,→ W . The variable 1 refers to the first value in the environment (counting from right to left); its
evaluation just returns that value.
η, W ` 1 ,→ W

fev 1

The meaning of an expression F ↑ in an environment η, W is the same as the meaning
of F in the environment η. Intuitively, the environment references from F into η
are shifted by one. The typical case is one where a reference to the nth value in η
is represented by the expression 1↑ · · · ↑, where the shift operator is applied n − 1
times.
η ` F ,→ W
fev ↑
η, W 0 ` F ↑ ,→ W
A functional abstraction usually immediately evaluates to itself. Here this is insufficient, since an expression ΛF may contain references to the environment η.
Thus we need to combine the environment η with ΛF to produce a closed (and
self-contained) value.
fev lam
η ` ΛF ,→ {η; ΛF }
In order to evaluate F1 F2 in an environment η we evaluate both F1 and F2 in that
environment, yielding the closure {η 0 ; ΛF10 } and value W2 , respectively. We then
add W2 to the environment η 0 , in effect binding the variable previously bound by
Λ in ΛF10 to W2 and then evaluate F10 in the extended environment to obtain the
overall value W .
η ` F1 ,→ {η 0 ; ΛF10 }

η ` F2 ,→ W2

η 0 , W2 ` F10 ,→ W

η ` F1 F2 ,→ W

fev app

Here is the implementation of this judgment as the type family feval in Elf.
feval : env -> exp’ -> val -> type.
%mode feval +N +F -W.
% Variables
fev_1 : feval (N , W) 1 W.
fev_^ : feval (N , W’) (F ^) W

%name feval D’.

<- feval N F W.

% Functions
fev_lam : feval N (lam’ F) (clo N (lam’ F)).
fev_app : feval N (app’ F1 F2) W
<- feval N F1 (clo N’ (lam’ F1’))
<- feval N F2 W2
<- feval (N’ , W2) F1’ W.
We have written this signature in a way that emphasizes its operational reading,
because it serves as an implementation of an interpreter. As an example, consider
the evaluation of the expression (Λ (Λ (1↑))) (Λ 1), which is a representation of
(lam x. lam y. x) (lam v. v).
?- D : feval empty (app’ (lam’ (lam’ (1 ^))) (lam’ 1)) W.
W = clo (empty , clo empty (lam’ 1)) (lam’ (1 ^)).
D’ = fev_app fev_lam fev_lam fev_lam.
The resulting closure, {(·, {·, Λ1}); Λ(1↑)}, represents the de Bruijn expressions
Λ(Λ1), since (1↑) refers to the first value in the environment.
The translation between ordinary and de Bruijn expressions is specified by the
following rules which employ a parametric and hypothetical judgment.
w⇔x
..
.
η ` F1 ↔ e1

η ` F2 ↔ e2

η ` F1 F2 ↔ e1 e2
W ⇔e
η, W ` 1 ↔ e

tr app

tr 1

u

η, w ` F ↔ e
η ` ΛF ↔ lam x. e

tr lamw,x,u

η`F ↔e
η, W ` F ↑ ↔ e

tr ↑

where the rule tr lam is restricted to the case where w and x are new parameters
not free in any other hypothesis, and u is a new label. The translation of values is
defined by a single rule in this language fragment.
η ` ΛF ↔ lam x. e
{η; ΛF } ⇔ lam x. e

vtr lam

As remarked earlier this translation can be non-deterministic if η and e are given
and F is to be generated. This is the direction in which this judgment would be
used for compilation. Here is an example of a translation.

u

·, w ` 1 ↔ x

tr 1

·, w, w0 ` 1↑ ↔ x

tr ↑
0

tr lamw, ,y,u

·, w ` Λ1↑ ↔ lam y. x

0

w,x,u

· ` ΛΛ1↑ ↔ lam x. lam y. x

tr lam

w00 ⇔ v

u00

·, w00 ` 1 ↔ v

tr 1

· ` Λ1 ↔ lam v. v

· ` (ΛΛ1↑) (Λ1) ↔ (lam x. lam y. x) (lam v. v)

tr lamw

00

,v,u00

tr app

The representation of the translation judgment relies on the standard technique
for representing deductions of hypothetical judgments as functions.
trans : env -> exp’ -> exp -> type. %name trans C.
vtrans : val -> exp -> type.
%name vtrans U.
% can be used in different directions
%mode trans +N +F -E.
%mode vtrans +W -V.
% Functions
tr_lam : trans N (lam’ F) (lam E)
<- ({w:val} {x:exp}
vtrans w x -> trans (N , w) F (E x)).
tr_app : trans N (app’ F1 F2) (app E1 E2)
<- trans N F1 E1
<- trans N F2 E2.
% Variables
tr_1 : trans (N , W) 1 E
<- vtrans W E.
tr_^ : trans (N , W) (F ^) E
<- trans N F E.
% Values
vtr_lam : vtrans (clo N (lam’ F)) (lam E)
<- trans N (lam’ F) (lam E).
The judgment
w⇔x
..
.

u

η, w ` F ↔ e

in the premiss of the tr lam is parametric in the variables w and x and hypothetical
in u. It is represented by a function which, when given a value W 0 , an expression
e0 , and a deduction U 0 :: W 0 ⇔ e0 returns a deduction D0 :: η, W 0 ` F ↔ [e0 /x]e.
This property is crucial in the proof of compiler correctness.
The signature above can be executed as a non-deterministic program for translation between de Bruijn and ordinary expressions in both directions. For the
compilation of expressions it is important to keep the clauses tr_1 and tr_^ in the
given order so as to avoid unnecessary backtracking. This non-determinism arises,
since the expression E in the rules tr_1 and tr_^ does not change in the recursive
calls. For other possible implementations see Exercise 6.3. Here is an execution
which yields the example deduction above.
?- C : trans empty F (app (lam [x] lam [y] x) (lam [z] z)).
F = app’ (lam’ (lam’ (1 ^))) (lam’ 1).
C =
tr_app (tr_lam ([w:val] [x:exp] [u3:vtrans w x] tr_1 u3))
(tr_lam
([w:val] [x:exp] [u1:vtrans w x]
tr_lam ([w1:val] [x1:exp] [u2:vtrans w1 x1]
tr_^ (tr_1 u1)))).
It is not immediately obvious that every source expression e can in fact be
compiled using this judgment. This is the subject of the following theorem.
Theorem 6.1 For every closed expression e there exists a de Bruijn expression F
such that · ` F ↔ e.
Proof: A direct attempt at an induction argument fails—a typical situation when
proving properties of judgments which involve hypothetical reasoning. However,
the theorem follows immediately from Lemma 6.2 below.
2
Lemma 6.2 Let w1 , . . . , wn be parameters ranging over values and let η be the
environment ·, wn, . . . , w1 . Furthermore, let x1 , . . . , xn range over expression variables. For any expression e with free variables among x1 , . . . , xn there exists a de
Bruijn expression F and a deduction C of η ` F ↔ e from hypotheses u1 :: w1 ⇔
x1 , . . . , un :: wn ⇔ xn .
Proof: By induction on the structure of e.
Case: e = e1 e2 . By induction hypothesis on e1 and e2 , there exist F1 and F2 and
deductions C1 :: η ` F1 ↔ e1 and C2 :: η ` F2 ↔ e2 . Applying the rule tr app
to C1 and C2 yields the desired deduction C :: η ` F1 F2 ↔ e1 e2 .

Case: e = lam x. e1 . Here we apply the induction hypothesis to the expression
e1 , environment η, w for a new parameter w, and hypotheses u1 :: w1 ⇔
x1 , . . . , un :: wn ⇔ xn , u :: w ⇔ x to obtain an F1 and a deduction
u
w⇔x
C1
η, w ` F1 ↔ e1
possibly also using hypotheses labelled u1 , . . . , un . Note that e1 is an expression with free variables among x1 , . . . , xn, x. Applying the rule tr lam
discharges the hypothesis u and we obtain the desired deduction

C=

u
w⇔x
C1
η, w ` F1 ↔ e1
η ` ΛF1 ↔ lam x. e1

tr lamu

Case: e = x. Then x = xi for some i between 1 and n and we let F = 1 ↑ · · · ↑
| {z }
i−1 times

and
wi ⇔ xi
C=

ui

·, wn, . . . , wi ` 1 ↔ xi
···

tr 1
tr ↑

·, wn, . . . , w1 ` 1↑ · · · ↑ ↔ xi

tr ↑

2
This proof cannot be represented directly in Elf because we cannot employ the
usual technique for representing hypothetical judgments as functions. The difficulty
is that the order of the hypotheses is important for returning the correct variable
1↑ · · · ↑, but hypothetical judgments are generally invariant under reordering of hypotheses. Hannan [Han91] has suggested a different, deterministic translation for
which termination is relatively easy to show, but which complicates the proofs of the
remaining properties of compiler correctness. Thus our formalization does not capture the desirable property that compilation always terminates. All the remaining
parts, however, are implemented. The first property states that translation followed
by evaluation leads to the same result as evaluation followed by translation. We
generalize this for arbitrary environments η in order to allow a proof by induction.

This property is depicted in the following diagram.
D :: e ,→ v

e
C :: η ` F ↔ e

-v
..6
..

?. . . . . . . . . . . . . . . . . -. .

η; F

D0 :: η ` F ,→ W

.. U :: W ⇔ v
..
W

The solid lines indicate deductions that are assumed, dotted lines represent the
deductions whose existence we assert and prove below.
Lemma 6.3 For any closed expressions e and v, environment η, de Bruijn expression F , deductions D :: e ,→ v and C :: η ` F ↔ e, there exist a value W and
deductions D0 :: η ` F ,→ W and U :: W ⇔ v.
Proof: By induction on the structures of D :: e ,→ v and C :: η ` F ↔ e. In
this induction we assume the induction hypothesis on the premisses of D and for
arbitrary C and on the premisses of C, but for the same D. This is sometimes called
lexicographic induction on the pair consisting of D and C. It should be intuitively
clear that this form of induction is valid. We represent this proof as a judgment
relating the four deductions involved in the diagram.
map_eval : eval E V -> trans N F E
-> feval N F W -> vtrans W V -> type.
%mode map_eval +D +C -D’ -U.
Case: C ends in an application of the tr 1 rule.
C=

U1
W1 ⇔ e
η1 , W1 ` 1 ↔ e

D :: e ,→ v
C1 :: η10 ` ΛF10 ↔ e and W1 = {η10 ; ΛF10 }
e = lam x. e1
v = lam x. e1 = e

tr 1
Assumption
By inversion on U1
By inversion on C1
By inversion on D

Then W = W1 , U = U1 :: W1 ⇔ e and D0 = fev 1 :: η1 , W1 ` 1 ,→ W1 satisfy
the requirements of the theorem. This case is captured in the clause
mp_1 : map_eval (ev_lam) (tr_1 (vtr_lam (tr_lam C2)))
(fev_1) (vtr_lam (tr_lam C2)).

Case: C ends in an application of the tr ↑ rule.
C=

C1
η 1 ` F1 ↔ e
η1 , W10 ` F1 ↑ ↔ e

D :: e ,→ v
D10 :: η1 ` F1 ,→ W1
and U1 :: W1 ⇔ v

tr ↑
Assumption
By ind. hyp. on D and C1

Now we let W = W1 , U = U1 , and obtain D0 :: η1 , W10 ` F1 ↑ ,→ W1 by fev ↑
from D10 .
mp_^ : map_eval D (tr_^ C1) (fev_^ D1’) U1
<- map_eval D C1 D1’ U1.
For the remaining cases we assume that the previous two cases do not apply. We
refer to this assumption as exclusion.
Case: D ends in an application of the ev lam rule.
D = lam x. e ,→ lam x. e ev lam
1
1
C :: η ` F ↔ lam x. e1
F = ΛF1

By assumption
By inversion and exclusion

Then we let W = {η; ΛF1}, D0 = fev lam :: η ` ΛF1 ,→ {η; ΛF1}, and obtain
U :: {η; ΛF1} ⇔ lam x. e1 by vtr lam from C.
mp_lam : map_eval (ev_lam) (tr_lam C1)
(fev_lam) (vtr_lam (tr_lam C1)).
Case: D ends in an application of the ev app rule.
D=

D1
e1 ,→ lam x. e01

D2
e2 ,→ v2

D3
[v2 /x]e01 ,→ v

ev app

e1 e2 ,→ v
This is the most interesting case, since it contains the essence of the argument how substitution can be replaced by binding variables to values in an
environment.

C :: η ` F ↔ e1 e2
F = F1 F2 ,
C1 :: η ` F1 ↔ e1 , and
C2 :: η ` F2 ↔ e2
D20 :: η ` F2 ,→ W2 and
U2 :: W2 ⇔ v2
D10 :: η ` F1 ,→ W1 and
U1 :: W1 ⇔ lam x. e01
W1 = {η1 ; ΛF10 } and
C10 :: η1 ` ΛF10 ↔ lam x. e01

By assumption

By inversion and exclusion
By ind. hyp. on D2 and C2
By ind. hyp. on D1 and C1
By inversion on U1

Applying inversion again to C10 shows that the premiss must be the deduction
of a hypothetical judgment. That is,
C10

=

u
w⇔x
C3
η1 , w ` F10 ↔ e01

where w is a new parameter ranging over values. This judgment is parametric
in w and x and hypothetical in u. We can thus substitute W2 for w, v2 for x,
and U2 for u to obtain a deduction
C30 :: η1 , W2 ` F10 ↔ [v2 /x]e01 .
Now we apply the induction hypothesis to D3 and C30 to obtain a W3 and
D30 :: η1 , W2 ` F10 ,→ W3 and
U3 :: W3 ⇔ v.
We let W = W3 , U = U3 , and obtain D0 :: η ` F1 F2 ,→ W by fev app from
D10 , D20 , and D30 .
The implementation of this relatively complex reasoning employs again the
magic of hypothetical judgments: the substitution we need to carry out to
obtain C30 from C3 is implemented as a function application.
mp_app : map_eval (ev_app D3 D2 D1) (tr_app C2 C1)
(fev_app D3’ D2’ D1’) U3
<- map_eval D1 C1 D1’ (vtr_lam (tr_lam C3))
<- map_eval D2 C2 D2’ U2
<- map_eval D3 (C3 W2 V2 U2) D3’ U3.
This completes the proof once we have convinced ourselves that all possible cases
have been considered. Note that whenever C ends in an application of the tr 1 or
tr ↑ rules, then the first two cases apply. Otherwise one of the other two cases must
apply, depending on the shape of D.
2

Theorem 6.1 and Lemma 6.3 together guarantee completeness of the translation.
Theorem 6.4 (Completeness) For any closed expressions e and v and evaluation
D :: e ,→ v, there exist a de Bruijn expression F , a value W and deductions C :: · `
F ↔ e, D0 :: · ` F ,→ W , and U :: W ⇔ v.
Proof: Lemma 6.3 shows that an evaluation D0 :: η ` F ,→ W and a translation
W ⇔ v exist for any translation C :: η ` F ↔ e. Theorem 6.1 shows that a
particular F and translation C :: · ` F ↔ e exist, thus proving the theorem.
2
Completeness is insufficient to guarantee compiler correctness. For example,
the translation of values W ⇔ v could relate any expression v to any value W ,
which would make the statement of the previous theorem almost trivially true. We
need to check a further property, namely that any value which could be produced
by evaluating the compiled code, could also be produced by direct evaluation as
specified by the natural semantics. This is shown in the diagram below.

-

D :: e ,→ v
e ...................... v
C :: η ` F ↔ e

?

η; F

6

-W

U :: W ⇔ v

D0 :: η ` F ,→ W

We call this property soundness of the compiler, since it prohibits the compiled code
from producing incorrect values. We prove this from a lemma which asserts the
existence of an expression v, evaluation D and translation U, given the translation
C and evaluation D0 . This yields the theorem by showing that the translation
U :: W ⇔ v, is uniquely determined from W .
Lemma 6.5 For any closed expression e, de Bruijn expression F , environment η,
value W , deductions D0 :: η ` F ,→ W and C :: η ` F ↔ e, there exist an expression
v and deductions D :: e ,→ v and U :: W ⇔ v.
Proof: The proof proceeds by a straightforward induction over the structure of
D0 :: η ` F ,→ W . It heavily employs inversion (as the proof of completeness,
Lemma 6.3). Interestingly, this proof can be implemented by literally the same
judgment. We leave it as exercise 6.6 to write out the informal proof—its representation from the proof of completeness is summarized below. Using is as a program
in this instance means that we assume that second and third arguments are given
and the first and last argument are logic variables whose instantiation terms are to
be constructed.

map_eval’ : eval E V -> trans N F E
-> feval N F W -> vtrans W V -> type.
%mode map_eval’ -D +C +D’ -U.
mp’_1 : map_eval’ (ev_lam) (tr_1 (vtr_lam (tr_lam C2)))
(fev_1) (vtr_lam (tr_lam C2)).
mp’_^ : map_eval’ D (tr_^ C1) (fev_^ D1’) U1
<- map_eval’ D C1 D1’ U1.
mp’_lam : map_eval’ (ev_lam) (tr_lam C1)
(fev_lam) (vtr_lam (tr_lam C1)).
mp’_app : map_eval’ (ev_app D3 D2 D1) (tr_app C2 C1)
(fev_app D3’ D2’ D1’) U3
<- map_eval’ D1 C1 D1’ (vtr_lam (tr_lam C3))
<- map_eval’ D2 C2 D2’ U2
<- map_eval’ D3 (C3 W2 V2 U2) D3’ U3.
%terminates D’ (map_eval’ _ C D’ _).

2
Theorem 6.6 (Uniqueness of Translations) For any value W if there exist a v
and a translation U :: W ⇔ v, then v and U are unique. Furthermore, for any
environment η and de Bruijn expression F , if there exist an e and a translation
C :: η ` F ↔ e, then e and C are unique.
Proof: By simultaneous induction on the structures of U and C. In each case, either
W or F uniquely determine the last inference. Since the translated expressions in
the premisses are unique by induction hypothesis, so is the translated value in the
conclusion.
2
The proof requires no separate implementation in Elf in the same way that
appeals to inversion remain implicit in the formulation of higher-level judgments.
It is obtained by direct inspection of properties of the inference rules.
Theorem 6.7 (Soundness) For any closed expressions e and v, de Bruijn expression F , environment η, value W , deductions D0 :: η ` F ,→ W , C :: η ` F ↔ e, and
U :: W ⇔ v, there exists a deduction D :: e ,→ v.
Proof: From Lemma 6.5 we infer the existence of a v, U, and D, given C and D0 .
Theorem 6.6 shows that v and U are unique, and thus the property must hold for
all v and U :: W ⇔ v, which is what we needed to show.
2

Adding Data Values and Recursion

In the previous section we treated only a very restricted core language of MiniML. In this section we will extend the compiler to the full Mini-ML language as
presented in Chapter 2. The main additions to the core language which affect the
compiler are data values (such as natural numbers and pairs) and recursion. The
language of de Bruijn expressions is extended by allowing constructors that parallel
ordinary expressions. We maintain a similar syntax, but mark de Bruijn expression
constructors with a prime (0 ).
Expressions

F

::= | z0 | s0 F | case0 F1 F2 F3
| hF1 , F2 i0 | fst0 F | snd0 F
| ΛF | F1 F2
| let0 val F1 in F2
| let0 name F1 in F2
| fix0 F
| 1 | F↑

Natural Numbers
Pairs
Functions
Definitions
Recursion
Variables

Expressions of the form F ↑ are not necessarily variables (where F is a sequence of
shifts applied to 1), but it may be intuitively helpful to think of them that way.
In the representation we need only first-order constants, since this language has no
constructs binding variables by name.
exp’

: type.

1
^
z’
s’
case’
pair’
fst’
snd’
lam’
app’
letv’
letn’
fix’

:
:
:
:
:
:
:
:
:
:
:
:
:

%name exp’ F f.

exp’.
exp’ ->
exp’.
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->
exp’ ->

exp’.
exp’.
exp’ ->
exp’ ->
exp’.
exp’.
exp’.
exp’ ->
exp’ ->
exp’ ->
exp’.

%postfix 20 ^.

exp’ -> exp’.
exp’.

exp’.
exp’.
exp’.

Next we need to extend the language of values. While data values can be
added in a straightforward fashion, let name and recursion present some difficulties.
Consider the evaluation rule for fixpoints.
[fix x. e/x]e ,→ v
fix x. e ,→ v

ev fix

We introduced the environment model of evaluation in order to eliminate the need
for explicit substitution, where an environment is a list of values. In the case of the
fixpoint construction we would need to bind the variable x to the expression fix x. e
in the environment in order to avoid substutition, but fix x. e is not a value. The
evaluation rules for de Bruijn expressions take advantage of the invariant that an
environment contains only values. In particular, the rule
η, W ` 1 ,→ W

fev 1

requires that an environment contain only values. We will thus need to add a
new environment constructor η + F in order to allow unevaluated expressions in the
environment. These considerations yield the following mutually recursive definitions
of environments and values. We mark data values with a star (∗ ) to distinguish them
from expressions and de Bruijn expressions with the same name.
Environments
η
Values W

::= · | η, W | η + F
::= | z∗ | s∗ W
| hW1 , W2 i∗
| {η; F }

Natural Numbers
Pairs
Closures

The Elf representation is direct.
env
val

: type.
: type.

%name env N.
%name val W w.

empty
,
+

: env.
: env -> val -> env.
: env -> exp’ -> env.

z*
s*

: val.
: val -> val.

pair*

: val -> val -> val.

clo

: env -> exp’ -> val.

%infix left 10 ,.
%infix left 10 +.

In the extension of the evaluation rule to this completed language, we must
exercise care in the treatment of the new environment constructor for unevaluated
expression: when such an expression is looked up in the environment, it must be
evaluated.
η ` F ,→ W
η + F ` 1 ,→ W

fev 1+

η ` F ,→ W
η + F 0 ` F ↑ ,→ W

fev ↑+

The rules involving data values generally follow the patterns established in the
natural semantics for ordinary expressions. The main departure from the earlier
formulation is the separation of values from expressions. We show only four of the
relevant rules.
η ` z0 ,→ z∗

η ` F ,→ W

fev z

η ` s0 F ,→ s∗ W

η ` F1 ,→ z∗

η ` F2 ,→ W

η ` case0 F1 F2 F3 ,→ W
η ` F1 ,→ s∗ W10

fev s

fev case z

η, W10 ` F3 ,→ W

η ` case0 F1 F2 F3 ,→ W

fev case s

Evaluating a let val-expression also binds a variable to value by extending the
environment.
η ` F1 ,→ W1
η, W1 ` F2 ,→ W
fev letv
η ` let val0 F1 in F2 ,→ W
Evaluating a let name-expression binds a variable to an expression and thus requires the new environment constructor.
η + F1 ` F2 ,→ W
η ` let name0 F1 in F2 ,→ W

fev letn

Fixpoint expressions are similar, except that the variable is bound to the fix expression itself.
η + fix0 F ` F ,→ W
fev fix
η ` fix0 F ,→ W
For example, fix x. x (considered on page 17) is represented by fix0 1. Intuitively,
evaluation of this expression should not terminate. An attempt to construct an
evaluation leads to the sequence
..
.
· ` fix0 1 ,→ W

fev fix

· +fix0 1 ` 1 ,→ W
· ` fix0 1 ,→ W

fev 1+
fev fix.

The implementation of these rules in Elf poses no particular difficulties. We
show only the rules from above.

feval : env -> exp’ -> val -> type.
%mode feval +N +F -W.

%name feval D.

% Variables
fev_1 : feval (N , W) 1 W.
fev_^ : feval (N , W’) (F ^) W
<- feval N F W.
fev_1+ : feval
<fev_^+ : feval
<-

(N + F) 1 W
feval N F W.
(N + F’) (F ^) W
feval N F W.

% Natural Numbers
fev_z : feval N z’ z*.
fev_s : feval N (s’ F) (s* W)
<- feval N F W.
fev_case_z : feval N (case’ F1 F2 F3) W
<- feval N F1 z*
<- feval N F2 W.
fev_case_s : feval N (case’ F1 F2 F3) W
<- feval N F1 (s* W1)
<- feval (N , W1) F3 W.
% Pairs
fev_pair : feval N (pair’ F1 F2) (pair* W1 W2)
<- feval N F1 W1
<- feval N F2 W2.
fev_fst : feval N (fst’ F) W1
<- feval N F (pair* W1 W2).
fev_snd : feval N (snd’ F) W2
<- feval N F (pair* W1 W2).
% Functions
fev_lam : feval N (lam’ F) (clo N (lam’ F)).
fev_app : feval N (app’ F1 F2) W
<- feval N F1 (clo N’ (lam’ F1’))
<- feval N F2 W2
<- feval (N’ , W2) F1’ W.
% Definitions
fev_letv : feval N (letv’ F1 F2) W

<- feval N F1 W1
<- feval (N , W1) F2 W.
fev_letn : feval N (letn’ F1 F2) W
<- feval (N + F1) F2 W.
% Recursion
fev_fix : feval N (fix’ F) W
<- feval (N + (fix’ F)) F W.

Next we need to extend the translation between expressions and de Bruijn expressions and values. We show a few interesting cases in the extended judgments
η ` F ↔ e and W ⇔ v. The case for let val is handled just like the case for lam,
since we will always substitute a value for the variable bound by the let during
execution.

0

η`z ↔z

tr z

η`F ↔e
η ` s0 F ↔ s e

w⇔x
..
.
η ` F1 ↔ e1
0

tr s

u

η, w ` F2 ↔ e2

η ` let val F1 in F2 ↔ let x = e1 in e2

tr letv w,x,u

where the right premiss of tr let is parametric in w and x and hypothetical in u.
In order to preserve the basic structure of the proofs of lemmas 6.3 and 6.5, we
must treat the let name and fix constructs somewhat differently: we extend the
environment with an expression parameter (not a value parameter) using the new

environment constructor +.
η`f ⇔x
..
.
η ` F1 ↔ e1

u

η + f ` F2 ↔ e2

η ` let name0 F1 in F2 ↔ let x = e1 in e2

η`f ↔x
..
.

u

η+f `F ↔e
η ` fix0 F ↔ fix x. e
η`F ↔e
η+F ` 1 ↔ e

tr fixf,x,u

η`F ↔e

tr 1+

tr letnf,x,u

η + F0 ` F↑ ↔ e

tr ↑+

Finally, the value translation does not have to deal with fixpoint-expressions (they
are not values). We only show the three new cases.
∗

z ⇔z

W ⇔v

vtr z

∗

s W ⇔sv

W1 ⇔ v1

W2 ⇔ v2
∗

hW1 , W2 i ⇔ hv1 , v2 i

vtr s

vtr pair

Deductions of parametric and hypothetical judgments are represented by functions,
as usual.
trans : env -> exp’ -> exp -> type. %name trans C.
vtrans : val -> exp -> type.
%name vtrans U.
% can be used in different directions
%mode trans +N +F -E.
%mode vtrans +W -V.
% Natural numbers
tr_z
: trans N z’ z.
tr_s
: trans N (s’ F) (s E)
<- trans N F E.

tr_case : trans N (case’ F1 F2 F3) (case E1 E2 E3)
<- trans N F1 E1
<- trans N F2 E2
<- ({w:val} {x:exp}
vtrans w x -> trans (N , w) F3 (E3 x)).
% Pairs
tr_pair : trans
<<tr_fst : trans
<tr_snd : trans
<-

N (pair’ F1 F2) (pair E1 E2)
trans N F1 E1
trans N F2 E2.
N (fst’ F1) (fst E1)
trans N F1 E1.
N (snd’ F1) (snd E1)
trans N F1 E1.

% Functions
tr_lam : trans N (lam’ F) (lam E)
<- ({w:val} {x:exp}
vtrans w x -> trans (N , w) F (E x)).
tr_app : trans N (app’ F1 F2) (app E1 E2)
<- trans N F1 E1
<- trans N F2 E2.
% Definitions
tr_letv: trans N (letv’ F1 F2) (letv E1 E2)
<- trans N F1 E1
<- ({w:val} {x:exp}
vtrans w x -> trans (N , w) F2 (E2 x)).
tr_letn: trans N (letn’ F1 F2) (letn E1 E2)
<- trans N F1 E1
<- ({f:exp’} {x:exp}
trans N f x -> trans (N + f) F2 (E2 x)).
% Recursion
tr_fix : trans N (fix’ F) (fix E)
<- ({f:exp’} {x:exp}
trans N f x -> trans (N + f) F (E x)).
% Variables
tr_1 : trans (N , W) 1 E <- vtrans W E.
tr_^ : trans (N , W) (F ^) E <- trans N F E.

tr_1+ : trans (N + F) 1 E <- trans N F E.
tr_^+ : trans (N + F’) (F ^) E <- trans N F E.
% Natural number values
vtr_z : vtrans z* z.
vtr_s : vtrans (s* W) (s V)
<- vtrans W V.
% Pair values
vtr_pair : vtrans (pair* W1 W2) (pair V1 V2)
<- vtrans W1 V1
<- vtrans W2 V2.
% Function values
vtr_lam : vtrans (clo N (lam’ F)) (lam E)
<- trans N (lam’ F) (lam E).
In order to extend the proof of compiler correctness in Section 6.1 we need to
extend various lemmas.
Theorem 6.8 For every closed expression e there exists a de Bruijn expression F
such that · ` F ↔ e.
Proof: We generalize analogously to Lemma 6.2 and prove the modified lemma by
induction on the structure of e (see Exercise 6.7).
2
Lemma 6.9 If W ⇔ e is derivable, then e Value is derivable.
Proof: By a straightforward induction on the structure of U :: W ⇔ e.

2

Lemma 6.10 If e Value and e ,→ v are derivable then e = v.
Proof: By a straightforward induction on the structure of P :: e Value.

2

The Elf implementations of the proofs of Lemmas 6.9 and 6.10 is straightforward
and can be found in the on-line material that accompanies these notes. The type
families are
vtrans_val : vtrans W E -> value E -> type.
%mode vtrans_val +U -P.
val_eval : value E -> eval E E -> type.
%mode val_eval +P -D.

The next lemma is the main lemma is the proof of completeness, that is, every
value which can obtained by direct evaluation can also be obtained by compilation,
evaluation of the compiled code, and translation of the returned value to the original
language.
Lemma 6.11 For any closed expressions e and v, environment η, de Bruijn expression F , deduction D :: e ,→ v and C :: η ` F ↔ e, there exist a value W and
deductions D0 :: η ` F ,→ W and U :: W ⇔ v.
Proof: By induction on the structure of D :: e ,→ v and C :: η ` F ↔ e. In this
induction, as in the proof of Lemma 6.3, we assume the induction hypothesis on the
premisses of D and for arbitrary C, and on the premisses of C if D remains fixed.
The implementation is an extension of the previous higher-level judgment,
map_eval : eval E V -> trans N F E
-> feval N F W -> vtrans W V -> type.
%mode map_eval +D +C -D’ -U.
We show only some of the typical cases—the others are straightforward and left to
the reader or remain unchanged from the proof of Lemma 6.3
Case: C ends in an application of the tr 1 rule.
C=

U1
W1 ⇔ e
η1 , W1 ` 1 ↔ e

tr 1

This case changes from the previous proof, since there we applied simple
inversion (there was only one possible kind of value) to conclude that e = v.
Here we need two lemmas from above.
D :: e ,→ v
P :: e Value
e=v

Assumption
By Lemma 6.9 from U1
By Lemma 6.10 from P

Hence we can let W be W1 , U be U1 , and D0 be fev 1 :: η1 , W1 ` 1 ,→ W1 .
The implementation explicitly appeals to the implementations of the lemmas.
mp_1 : map_eval D (tr_1 U1) (fev_1) U1
<- vtrans_val U1 P
<- val_eval P D.
Case: C ends in an application of the tr ↑ rule. This case proceeds as before.

mp_^ : map_eval D (tr_^ C1) (fev_^ D1’) U1
<- map_eval D C1 D1’ U1.

Case: C ends in an application of the tr 1+ rule.
C=

C1
η 1 ` F1 ↔ e
η 1 + F1 ` 1 ↔ e

tr 1+

D :: e ,→ v
D10 :: η1 ` F1 ,→ W1 and
U1 :: W1 ⇔ v
D0 :: η1 + F1 ` 1 ,→ W

Assumption
By ind. hyp. on D and C1
By fev 1+ from D10

and we can let W = W1 and U = U1 .
mp_1+ : map_eval D (tr_1+ C1) (fev_1+ D1’) U1
<- map_eval D C1 D1’ U1.
Case: C ends in an application of the tr ↑+ rule. This case is just like the tr ↑ case.
mp_^+ : map_eval D (tr_^+ C1) (fev_^+ D1’) U1
<- map_eval D C1 D1’ U1.
For the remaining cases we may assume that none of the four cases above apply.
We only show the case for fixpoints.
Case: D ends in an application of the ev fix rule.
D=

D1
[fix x. e1 /x]e1 ,→ v
fix x. e1 ,→ v

C :: η ` F ↔ fix x. e1

ev fix
By assumption

By inversion and exclusion (of the previous cases), C must end in an application of the tr fix rule and thus F = fix0 F1 for some F1 and there is a deduction
C1 , parametric in f and x and hypothetical in u, of the form
u
η`f ↔x
C1
η + f ` F1 ↔ e1

In this deduction we can substitute fix0 F1 for f and fix x. e1 for x, and
replace the resulting hypothesis u :: η ` fix0 F1 ↔ fix x. e1 by C! This way
we obtain a deduction
C10 :: η + fix0 F1 ` F1 ↔ [fix x. e1 /x]e1 .
Now we can apply the induction hypothesis to D1 and C10 which yields a W1
and deductions
D10 :: η + fix0 F1 ` F1 ,→ W1 and
U1 :: W1 ⇔ v

By ind. hyp. on D1 and C10

Applying fev fix to D10 results in a deduction
D0 :: η ` fix0 F1 ,→ W1
and we let W be W1 and U be U1 . In Elf, the substitutions into the hypothetical deduction are implemented by applications of the representing function
C1.
mp_fix : map_eval (ev_fix D1) (tr_fix C1)
(fev_fix D1’) U1
<- map_eval D1 (C1 (fix’ F1) (fix E1) (tr_fix C1))
D1’ U1.

2
This lemma and the totality of the translation relation in its expression argument
(Theorem 6.8) together guarantee completeness of the translation.
Theorem 6.12 (Completeness) For any closed expressions e and v and evaluation
D :: e ,→ v, there exist a de Bruijn expression F , a value W and deductions C :: · `
F ↔ e, D0 :: · ` F ,→ W , and U :: W ⇔ v.
Proof: As in the proof of Theorem 6.4, but using Lemma 6.11 and Theorem 6.8
instead of Lemma 6.3 and Theorem 6.1.
2
Lemma 6.13 For any closed expression e, de Bruijn expression F , environment η,
value W , deduction D0 :: η ` F ,→ W and C :: η ` F ↔ e, there exist an expression
v and deductions D :: e ,→ v and U :: W ⇔ v.
Proof: By induction on the structure of D0 :: η ` F ,→ W . The family map_eval
which implements the main lemma in the soundness proof, also implements the
proof of this lemma without any change.
2

Theorem 6.14 (Uniqueness of Translations) For any value W if there exist a v
and a translation U :: W ⇔ v, then v and U are unique. Furthermore, for any
environment η and de Bruijn expression F , if there exist an e and a translation
C :: η ` F ↔ e, then e and C are unique.
Proof: As before, by a simultaneous induction on the structures of U and C.

2

Theorem 6.15 (Soundness) For any closed expressions e and v, de Bruijn expression F , environment η, value W , deductions D0 :: η ` F ,→ W , C :: η ` F ↔ e, and
U :: W ⇔ v, there exists a deduction D :: e ,→ v.
Proof: From Lemma 6.13 we infer the existence of a v, U, and D, given C and D0 .
Theorem 6.14 shows that v and U are unique, and thus the property must hold for
2
all v and U :: W ⇔ v, which is what we needed to show.

6.3

Computations as Transition Sequences

So far, we have modelled evaluation as the construction of a deduction of the evaluation judgment. This is true for evaluation based on substitution in Section 2.3
and for evaluation based on environments in Section 6.1. In an abstract machine
(and, of course, in an actual machine) a more natural model for computation is a
sequence of states. In this section we will develop the CLS machine, an abstract
machine similar in scope to the SECD machine [Lan64]. The CLS machine still interprets expressions, so the step from environment based evaluation to this abstract
machine does not involve any compilation. Instead, we flatten evaluation trees to
sequences of states that describe the computation. This flattening involves some
rather arbitrary decisions about which subcomputations should be performed first.
We linearize the evaluation deductions beginning with the deduction of the leftmost
premiss.
Throughout the remainder of this chapter, we will drop the prime (0 ) from the
expression constructors. This should not lead to any confusion, since we no longer
need to refer to the original expressions. Now consider the rule for evaluating pairs
as a simple example where an evaluation tree has two branches.
η ` F1 ,→ W1

η ` F2 ,→ W2

η ` hF1 , F2i ,→ hW1 , W2 i∗

fev pair

An abstract machine would presumably start in a state where it is given the
environment η and the expression hF1 , F2 i. The final state of the machine should
somehow indicate the final value hW1 , W2 i∗ . The computation naturally decomposes
into three phases: the first phase computes the value of F1 in environment η, the
second phase computes the value of F2 in environment η, and the third phase

combines the two values to form a pair. These phases mean that we have to preserve
the environment η and also the expression F2 while we are computing the value of F1 .
Similarly, we have to save the value W1 while computing the value of F2 . A natural
data structure for saving components of a state is a stack. The considerations
above suggest three stacks: a stack Ξ of environments, a stack of expressions to be
evaluated, and a stack S of values. However, we also need to remember that, after
the evaluation of F2 we need to combine W1 and W2 into a pair. Thus, instead
of a stack of expression to be evaluated, we maintain a program which consists of
expressions and special instructions (such as: make a pair written as mkpair ).
We will need more instructions later, but so far we have:
Instructions
I
Programs P
Environment Stacks Ξ
Value Stacks S
State St

::=
::=
::=
::=
::=

F | mkpair | . . .
done | I & P
· | Ξ; η
· | S, W
hΞ, P, Si

Note that value stacks are simply environments, so we will not formally distinguish
them from environments. The instructions of a program a sequenced with &; the
program done indicates that there are no further instructions, that is, computation
should stop.
A state consists of an environment stack Ξ, a program P and a value stack S,
written as hΞ, P, Si. We have single-step and multi-step transition judgments:
St =⇒ St 0
∗
St =⇒ St 0

St goes to St 0 in one computation step
St goes to St 0 in zero or more steps

We define the transition judgment so that
∗

h(·; η), F & done, ·i =⇒ h·, done, (·, W )i
corresponds to the evaluation of F in environment η to value W . The free variables
of F are therefore bound in the innermost environment, and the value resulting
from evaluation is deposited on the top of the value stack, which starts out empty.
Global evaluation is expressed in the judgment
∗

η ` F =⇒
=⇒ W

F computes to W in environment η

which is defined by the single inference rule
∗

h(·; η), F & done, ·i =⇒ h·, done, (·, W )i
∗

run.

η ` F =⇒
=⇒ W
∗

We prove in Theorem 6.19 that η ` F =⇒
=⇒ W iff η ` F ,→ W . We cannot prove this
statement directly by induction (in either direction), since during a computation

situations arise where the environment stack consists of more than a single environment, the remaining program is not done, etc. In one direction we generalize it
to
∗
h(Ξ; η), F & P, Si =⇒ hΞ, P, (S, W )i
if η ` F ,→ W . This is the subject of Lemma 6.16. A slightly modified form of the
converse is given in Lemma 6.18.
The transition rules and the remaining instructions can be developed systematically from the intuition provided above. First, we reconsider the evaluation of
pairing. The first rule decomposes the pair expression and saves the environment η
on the environment stack.
c pair :: h(Ξ; η), hF1, F2 i & P, Si =⇒ h(Ξ; η; η), F1 & F2 & mkpair & P, Si
Here c pair labels the rule and can be thought of as the deduction of the given
transition judgment. The evaluation of F1 , if it terminates, leads to a state
h(Ξ; η), F2 & mkpair & P, (S, W1 )i,
and the further evaluation of F2 then leads to a state
hΞ, mkpair & P, (S, W1 , W2 )i.
Thus, the mkpair instruction should cause the machine to create a pair from the
first two elements on the value stack and deposit the result again on the value stack.
That is, we need as another rule:
c mkpair :: hΞ, mkpair & P, (S, W1 , W2 )i =⇒ hΞ, P, (S, hW1 , W2 i∗ )i.
We consider one other construct in detail: application. To evaluate an application F1 F2 we first evaluate F1 and then we evaluate F2 . If the value of F1 is a
closure, we have to bind its variable to the value of F2 and continue evaluation in
an extended environment. The instruction that unwraps the closure and extends
the environment is called apply.
c app :: h(Ξ; η), F1 F2 & P, Si =⇒ h(Ξ; η; η), F1 & F2 & apply & P, Si
c apply :: hΞ, apply & P, (S, {η 0 ; ΛF10 }, W2 )i =⇒ h(Ξ; (η 0 , W2 )), F10 & P, Si
The rules for applying zero and successor are straightforward, but they necessitate
a new operator add1 to increment the first value on the stack.
cz
cs
c add1

:: h(Ξ; η), z & P, Si =⇒ hΞ, P, (S, z∗ )i
:: h(Ξ; η), s F & P, Si =⇒ h(Ξ; η), F & add1 & P, Si
:: hΞ, add1 & P, (S, W )i =⇒ hΞ, P, (S, s∗ W )i

For expressions of the form case F1 F2 F3 , we need to evaluate F1 and then evaluate
either F2 or F3 , depending on the value of F1 . This requires a new instruction,

branch, which either goes to the next instructions or skips the next instruction.
In the latter case it also needs to bind a new variable in the environment to the
predecessor of the value of F1 .
c case :: h(Ξ; η), case F1 F2 F3 & P, Si
=⇒ h(Ξ; η; η), F1 & branch & F2 & F3 & P, Si
c branch z :: h(Ξ; η), branch & F2 & F3 & P, (S, z∗ )i =⇒ h(Ξ; η), F2 & P, Si
c branch s :: h(Ξ; η), branch & F2 & F3 & P, (S, s∗ W )i
=⇒ h(Ξ; (η, W )), F3 & P, Si
Rules for fst and snd require new instructions to extract the first or second
element of the value on the top of the stack.
c fst
c getfst
c snd
c getsnd

::
::
::
::

h(Ξ; η), fst F & P, Si =⇒ h(Ξ; η), F & getfst & P, Si
hΞ, getfst & P, (S, hW1 , W2 i∗ )i =⇒ hΞ, P, (S, W1)i
h(Ξ; η), snd F & P, Si =⇒ h(Ξ; η), F & getsnd & P, Si
hΞ, getsnd & P, (S, hW1 , W2 i∗ )i =⇒ hΞ, P, (S, W2)i

In order to handle let val we introduce another new instruction bind , even
though it is not strictly necessary and could be simulated with other instructions
(see Exercise 6.10).
c let :: h(Ξ; η), let F1 in F2 & P, Si =⇒ h(Ξ; η; η), F1 & bind & F2 & P, Si
c bind :: h(Ξ; η), bind & F2 & P, (S; W1 )i =⇒ h(Ξ; (η, W1 )), F2 & P, Si
We leave the rules for recursion to Exercise 6.11. The rules for variables and
abstractions thus complete the specification of the single-step transition relation.
:: h(Ξ; (η, W )), 1 & P, Si =⇒ hΞ, P, (S, W )i
:: h(Ξ; (η, W 0 )), F ↑ & P, Si =⇒ h(Ξ; η), F & P, Si
:: h(Ξ; η), ΛF & P, Si =⇒ hΞ, P, (S, {η; ΛF })i

c1
c↑
c lam

The set of instructions extracted from these rules is
Instructions I

::= F | add1 | branch | mkpair | getfst | getsnd | apply | bind .

We view each of the transition rules for the single-step transition judgment as
an axiom. Note that there are no other inference rules for this judgment. A partial
computation is defined as a multi-step transition. This is easily defined via the
following two inference rules.
∗

St =⇒ St

id

∗

St =⇒ St 0

St 0 =⇒ St 00
∗

St =⇒ St

00

step

This definition guarantees that the end state of one transition matches the beginning
state of the remaining transition sequence. Without the aid of dependent types we

would have to define a computation as a list states and ensure externally that the
end state of each transition matches the beginning state of the next. This use of
dependent types to express complex constraints is one of the reasons why simple
lists do not arise very frequently in Elf programming.
∗
Deductions of the judgment St =⇒ St 0 have a very simple form: They all consist
of a sequence of single steps terminated by an application of the id rule. We will
follow standard practice and use a linear notation for sequences of steps:
St 1 =⇒ St 2 =⇒ · · · =⇒ St n
Similarly, we will mix multi-step and single-step transitions in sequences, with the
obvious meaning. We write C1 ◦ C2 for the result of appending computations C1 and
C2 . This only makes sense if the final state of C1 is the same as the start state of
C2 . The ◦ operator is associative (see Exercise 6.12).
Recall that a complete computation was defined as a sequence of transitions
from an initial state to a final state. The latter is characterized by the program
done, and empty environment stack, and a value stack containing exactly one value,
namely the result of the computation.
∗

h(·; η), F & done, ·i =⇒ h·, done, (·, W )i
∗

run

η ` F =⇒
=⇒ W
The representation of the abstract machine and the computation judgments
present no particular difficulties. We begin with the syntax.
instruction
program
envstack
state

:
:
:
:

type.
type.
type.
type.

%name
%name
%name
%name

instruction I.
program P.
envstack Ns.
state St.

% Instructions
ev
: exp’ -> instruction.
add1
: instruction.
branch : instruction.
mkpair : instruction.
getfst : instruction.
getsnd : instruction.
apply : instruction.
bind
: instruction.
% Programs
done : program.
&
: instruction -> program -> program.

%infix right 10 &.
% Environment stacks
emptys : envstack.
;
: envstack -> env -> envstack.
%infix left 10 ;.
% States
st : envstack -> program -> env -> state.
The computation rules are also a straightforward transcription of the rules above.
∗
The judgment St =⇒ St 0 is represented by a type St => St’ where => is a type
family indexed by two states and written in infix notation. We show only three
example rules.
=> : state -> state -> type.
%infix none 10 =>.
%mode => +St -St’.
c_z

%name => R.

: st (Ns ; N) (ev z’ & P) S => st Ns P (S , z*).

c_app
c_apply

:
=>
:
=>

st
st
st
st

(Ns ; N) (ev (app’ F1 F2) & P) S
(Ns ; N ; N) (ev F1 & ev F2 & apply & P) S.
Ns (apply & P) (S , clo N’ (lam’ F1’) , W2)
(Ns ; (N’ , W2)) (ev F1’ & P) S.

The multi-step transition is defined by the transcription of its two inference
rules. We write ~ in infix notation rather than step since it leads to a concise and
readable notation for sequences of computation steps.
=>* : state -> state -> type.
%infix none 10 =>*.
% no mode---this is not operational.
id

: St =>* St.

~

: St => St’
-> St’ =>* St’’
-> St =>* St’’.

%name =>* C.

%infix right 10 ~.
Complete computations appeal directly to the multi-step computation judgment.
∗
We write ceval K F W for η ` F =⇒
=⇒ W .

ceval : env -> exp’ -> val -> type.
% no mode---this is not operational
run :

st (emptys ; N) (ev F & done) (empty)
=>* st (emptys) (done) (empty , W)
-> ceval N F W.

While this representation is declaratively adequate it has a serious operational defect
when used for evaluation, that is, when η and F are given and W is to be determined.
The declaration for step (written as ~) solves the innermost subgoal first, that is
∗
we reduce the goal of finding a computation C 00 :: St =⇒ St 00 to finding a state St 0
0 ∗
00
0
and computation of C :: St =⇒ St and only then a single transition R :: St =⇒
St 0 . This leads to non-termination, since the interpreter is trying to work its way
backwards through the space of possible computation sequences. Instead, we can
get linear, backtracking-free behavior if we first find the single step R :: St =⇒ St 0
∗
and then the remainder of the computation C 0 :: St 0 =⇒ St 00 . Since there is exactly
one rule for any instruction I and id will apply only when the program P is done,
finding a computation now becomes a deterministic process. Executable versions
of the last two judgments are given below. They differ from the one above only in
the order of the recursive calls and it is a simple matter to relate the two versions
formally.
>=>* : state -> state -> type.
%infix none 10 >=>*.
%mode >=>* +St -St’.
id<
<=<

: St >=>* St.
: St >=>* St’’
<- St => St’
<- St’ >=>* St’’.
%infix left 10 <=<.
>ceval : env -> exp’ -> val -> type.
%mode >ceval +N +F -W.
>run

: >ceval N F W
<- st (emptys ; N) (ev F & done) (empty)
>=>* st (emptys) (done) (empty , W).

This example clearly illustrates that Elf should be thought of a uniform language in
which one can express specifications (such as the computations above) and implementations (the operational versions below), but that many specifications will not
be executable. This is generally the situation in logic programming languages.

In the informal development it is clear (and not usually separately formulated
as a lemma) that computation sequences can be concatenated if the final state of
the first computation matches the initial state of the second computation. In the
formalization of the proofs below, we will need to explicitly implement a type family
that appends computation sequences. It cannot be formulated as a function, since
such a function would have to be recursive and is thus not definable in LF.
append : st Ns P S =>* st Ns’ P’ S’
-> st Ns’ P’ S’ =>* st Ns’’ P’’ S’’
-> st Ns P S =>* st Ns’’ P’’ S’’
-> type.
%mode append +C +C’ -C’’.
The defining clauses are left as Exercise 6.12.
We now return to the task of proving the correctness of the abstract machine.
The first lemma states the fundamental motivating property for this model of computation.
Lemma 6.16 Let η be an environment, F an expression, and W a value such that
η ` F ,→ W . Then, for any environment stack Ξ, program P and stack S,
∗

h(Ξ; η), F & P, Si =⇒ hΞ, P, (S, W )i
Proof: By induction on the structure of D :: η ` F ,→ W . We will construct a
∗
deduction of C :: h(Ξ; η), F & P, Si =⇒ hΞ, P, (S, W )i. The proof is straightforward
and we show only two typical cases. The implementation in Elf takes the form of a
higher-level judgment subcomp that relates evaluations to computation sequences.
subcomp : feval N F W
-> st (Ns ; N) (ev F & P) S =>* st Ns P (S , W)
-> type.
%mode +{N:env} +{F:exp’} +{W:val} +{Ns:envstack} +{P:program} +{S:env}
+{D:feval N F W} -{C:st (Ns ; N) (ev F & P) S =>* st Ns P (S , W)}
subcomp D C.
Note that the mode declaration here is in the “full” form in order to express that the
implicitly quantified variables Ns, P, and S are input arguments rather than output
arguments. The system would have inferred the latter from the simpler declaration
%mode subcomp +D -C.
Case: D ends in an application of the rule fev z.
D = η ` z ,→ z∗ fev z.

178

CHAPTER 6. COMPILATION
Then the single-step transition
∗

h(Ξ; η), z & P, Si =⇒ hΞ, P, (S, z∗ )i
satisfies the requirements of the lemma. The clause corresponding to this case:
sc_z : subcomp (fev_z) (c_z ~ id).
Case: D ends in an application of the fev app rule.
D=

D2
D1
η ` F1 ,→ {η 0 ; ΛF10 } η ` F2 ,→ W2

D3
η 0 , W2 ` F10 ,→ W

η ` F1 F2 ,→ W

fev app

Then
h(Ξ; η), F1 F2 & P, Si
=⇒ h(Ξ; η; η), F1 & F2 & apply & P, Si
∗
=⇒ h(Ξ; η), F2 & apply & P, (S, {η 0 ; ΛF10 })i
∗
=⇒ hΞ, apply & P, (S, {η 0 ; ΛF10 }, W2 )i
=⇒ h(Ξ; (η 0 , W2 )), F10 & P, Si
∗
=⇒ hΞ, P, (S, W )i

By rule c app
By ind. hyp. on D1
By ind. hyp. on D2
By rule c apply
By ind. hyp. on D3 .

The implementation of this case requires the append family defined above.
Note how an appeal to the induction hypothesis is represented as a recursive
call.
sc_app : subcomp (fev_app D3 D2 D1) C
<- subcomp D1 C1
<- subcomp D2 C2
<- subcomp D3 C3
<- append (c_app ~ C1) C2 C’
<- append C’ (c_apply ~ C3) C.

2
The first direction of Theorem 6.19 is a special case of this lemma. The other
direction is more intricate. The basic problem is to extract a tree-structured evaluation from a linear computation. We must then show that this extraction will always
succeed for complete computations. Note that it is obviously not possible to extract evaluations from arbitrary incomplete sequences of transitions of the abstract
machine.

In order to write computation sequences more concisely, we introduce some
∗
notation. Let R :: St =⇒ St 0 and C :: St 0 =⇒ St 00 . Then we write
R ∼ C :: St =⇒ St 00
for the computation which begins with R and then proceeds with C. Such a computation exists by the step inference rule. This corresponds directly to the notation
in the Elf implementation.
For the proof of the central lemma of this section, we will need a new form
of induction often referred to as complete induction. During a proof by complete
induction we assume the induction hypothesis not only for the immediate premisses
of the last inference rule, but for all proper subderivations. Intuitively, this is
justified, since all proper subderivations are “smaller” than the given derivation.
For a more formal discussion of the complete induction principle for derivations see
Section 6.4. The judgment C < C 0 (C is a proper subcomputation of C 0 ) is defined
by the following inference rules.

C<R∼C

C < C0

sub imm

C < R ∼ C0

sub med

It is easy to see that the proper subcomputation relation is transitive.
Lemma 6.17 If C1 < C2 and C2 < C3 then C1 < C3 .
Proof: By a simple induction (see Exercise 6.12).

2

The implementation of this ordering and the proof of transitivity are immediate.
< : (st Ns1 P1 S1) =>* (st Ns P S)
-> (st Ns2 P2 S2) =>* (st Ns P S)
-> type.
%infix none 8 <.
sub_imm : C < R ~ C.
sub_med : C < C’
-> C < R ~ C’.
The representation of the proof of transitivity is left to Exercise 6.12.
We are now prepared for the lemma that a complete computation with an appropriate initial state can be translated into an evaluation followed by another complete
computation.
Lemma 6.18 If
∗

C :: h(Ξ, η), F & P, Si =⇒ h·, done, (·, W 0 )i

there exists a value W , an evaluation
D :: η ` F ,→ W,
and a computation
∗

C 0 :: hΞ, P, (S, W )i =⇒ h·, done, (·, W 0)i
such that C 0 < C.
Proof: By complete induction on the C. We only show a few cases; the others
similar. We use the abbreviation final = h·, done, (·, W 0)i. The representing type
family spl is indexed by three deductions: C, C 0 , and D. We do not explicitly
represent the derivation which shows that C 0 < C since the Twelf system can check
this automatically using a %reduces declaration, placed after all rules for spl.
spl : (st (Ns ; N) (ev F & P) S) =>* (st emptys done (empty , W’))
-> feval N F W
-> (st Ns P (S , W)) =>* (st emptys done (empty , W’))
-> type.
%mode spl +C -D -C’.
... clauses for spl ...
%reduces C’ < C (spl C _ C’).
%terminates C (spl C _ _).
Now back to the proof.
Case: C begins with c z, that is, C = c z ∼ C1 . Then W = z∗ ,
D = K ` z ,→ z∗

fev z,

and C 0 = C1 . Furthermore, C 0 = C1 < c z ∼ C1 by rule sub imm. The
representation in Elf:
spl_z : spl (c_z ~ C1) (fev_z) C1.
Case: C begins with c app. Then C = c app ∼ C1 where
∗

C1 :: h(Ξ; η; η), F1 & F2 & apply & P, Si =⇒ final .
By induction hypothesis on C1 there exists a W1 , an evaluation
D1 :: η ` F1 ,→ W1
and a computation
∗

C2 :: h(Ξ; η), F2 & apply & P, (S, W1 )i =⇒ final

such that C2 < C1 . We can thus apply the induction hypothesis to C2 to obtain
a W2 , an evaluation
D2 :: η ` F2 ,→ W2
and a computation
∗

C3 :: hΞ, apply & P, (S, W1 , W2 )i =⇒ final
such that C3 < C2 . By inversion, C3 = c apply ∼ C30 and W1 = {η 0 ; ΛF10 } where
∗

C30 :: h(Ξ; (η 0 , W2 )), F10 & P, Si =⇒ final .
Then C30 < C3 and by induction hypothesis on C30 there is a value W3 , an
evaluation
D3 :: η 0 , W2 ` F10 ,→ W3
and a compuation

∗

C4 :: hΞ, P, (S, W3)i =⇒ final .
Now we let W = W3 and we construct D :: η ` F1 F2 ,→ W3 by an application
of the rule fev app to the premisses D1 , D2 , and D3 . Furthermore we let C 0 =
C4 and conclude by some elementary reasoning concerning the subcomputation
relation that C 0 < C.
The representation of this subcase of this case requires three explicit appeals
to the transitivity of the subcomputation ordering. In order to make this at
all intelligible, we use the name C2<C1 (one identifier) for the derivation that
C2 < C1 and similarly for other such derivations.
spl_app : spl (c_app ~ C1)
(fev_app D3 D2 D1) C4
<- spl C1 D1 C2
<- spl C2 D2 (c_apply ~ C3’)
<- spl C3’ D3 C4.

2
Now we have all the essential lemmas to prove the main theorem.
∗

Theorem 6.19 η ` F ,→ W is derivable iff η ` F =⇒
=⇒ W is derivable.
∗

=⇒ W iff there is a computation
Proof: By definition, η ` F =⇒
∗

C :: h(·; η), F & done, ·i =⇒ h·, done, (·, W )i.
One direction follows immediately from Lemma 6.16: if η ` F ,→ W then
∗

h(Ξ; η), F & P, Si =⇒ hΞ, P, (S, W )i
for any Ξ, P , and S and in particular for Ξ = ·, P = done and S = ·. The
implementation of this direction in Elf:

cev_complete : feval N F W -> ceval N F W -> type.
%mode cev_complete +D -C.
cevc : cev_complete D (run C) <- subcomp D C.
%terminates [] (cev_complete D _).
For the other direction, assume there is a deduction C of the form shown above.
By Lemma 6.18 we know that there exist a W 0 , an evaluation
D0 :: η ` F ,→ W 0
and a computation
∗

C 0 :: h·, done, (·, W 0 )i =⇒ h·, done, (·, W )i
such that C 0 < C. Since there is no transition rule for the program done, C 0 must
be id and W = W 0 . Thus D = D0 fulfills the requirements of the theorem. This is
implemented as follows.
cls_sound : ceval N F W -> feval N F W -> type.
%mode cls_sound +C -D.
clss : cls_sound (run C) D <- spl C D (id).
%terminates [] (cls_sound C _).

2
6.4

Complete Induction over Computations

Here we briefly justify the principle of complete induction used in the proof of
Lemma 6.18. We repeat the definition of proper subcomputations and also define
a general subcomputation judgment which will be useful in the proof.
C < C0
C ≤ C0

C is a proper subcomputation of C 0 , and
C is a subcomputation of C 0 .

These judgments are defined via the following inference rules.

C<R∼C
C < C0
C ≤ C0

sub imm

leq sub

C < C0
C < R ∼ C0
C ≤C

sub med

leq eq

We only need one simple lemma regarding the subcomputation judgment.

Lemma 6.20 If C ≤ C 0 is derivable, then C < R ∼ C 0 is derivable.
Proof: By analyzing the two possibilities for the deduction of the premiss and
constructing an immediate deduction for the conclusion in each case.
2
We call a property P of computations complete if it satisfies:
For every C, the assumption that P holds for all C 0 < C implies that P
holds for C.
Theorem 6.21 (Principle of Complete Induction over Computations) If a property
P of computations is complete, then P holds for all computations.
Proof: We assume that P is complete and then prove by ordinary structural induction that for every C and for every C 0 ≤ C, P holds of C.
Case: C = id . By inversion, there is no C 0 such that C 0 < id. Thus P holds for all
C 0 < id. Since P is complete, this implies that P holds for id.
Case: C = R ∼ C1 . The induction hypothesis states that for every C10 ≤ C1 , P
holds of C10 . We have to show that for every C2 ≤ R ∼ C1 , property P holds
of C2 . By inversion, there are two subcases, depending on the evidence for
C2 ≤ R ∼ C1 .
Subcase: C2 = R ∼ C1 . The induction hypothesis and Lemma 6.20 yield
that for every C10 < R ∼ C1 , P holds of C1 . Since P is complete, P must
thus hold for R ∼ C1 = C2 .
Subcase: C2 < R ∼ C1 . Then by inversion either C1 = C2 or C1 < C2 . In
either case C2 ≤ C1 by one inference. Now we can apply the induction
hypothesis to conclude that P holds of C2 .

2
6.5

A Continuation Machine

The natural semantics for Mini-ML presented in Chapter 2 is called a big-step
semantics, since its only judgment relates an expression to its final value—a “big
step”. There are a variety of properties of a programming language which are
difficult or impossible to express as properties of a big-step semantics. One of the
central ones is that “well-typed programs do not go wrong”. Type preservation,
as proved in Section 2.6, does not capture this property, since it presumes that we
are already given a complete evaluation of an expression e to a final value v and
then relates the types of e and v. This means that despite the type preservation
theorem, it is possible that an attempt to find a value of an expression e leads to an

intermediate expression such as fst z which is ill-typed and to which no evaluation
rule applies. Furthermore, a big-step semantics does not easily permit an analysis
of non-terminating computations.
An alternative style of language description is a small-step semantics. The main
judgment in a small-step operational semantics relates the state of an abstract
machine (which includes the expression to be evaluated) to an immediate successor
state. These small steps are chained together until a value is reached. This level of
description is usually more complicated than a natural semantics, since the current
state must embody enough information to determine the next and all remaining
computation steps up to the final answer. It is also committed to the order in
which subexpressions are evaluated and thus somewhat less abstract than a natural,
big-step semantics.
In this section we construct a machine directly from the original natural semantics of Mini-ML in Section 2.3 (and not from the environment-based semantics in
Section 6.1). This illustrates the general technique of continuations to sequentialize computations. Another application of the technique at the level of expressions
(rather than computations) is given in Section ??.
Our goal now is define a small-step semantics. For this, we isolate an expression
e to be evaluated, and a continuation K which contains enough information to carry
out the rest of the evaluation necessary to compute the overall value. For example,
to evaluate a pair he1 , e2 i we first compute the value of e1 , remembering that the
next task will be the evaluation of e2 , after which the two values have to be paired.
This also shows the need for intermediate instructions, such as “evaluate the second
element of a pair” or “combine two values into a pair”. One particular kind of
instruction, written as ev e, triggers the first step in the computation based on the
structure of e.
Because we always fully evaluate one expression before moving on to the next,
the continuation has the form of a stack. Because the result of evaluating the current
expression must be communicated to the continuation, each item on the stack is a
function from values to instructions. Finally, when we have computed a value, we
return it by applying the first item on the continuation stack. Thus the following
structure emerges, to be supplement by further auxiliary instructions as necessary.
Instructions
i ::= ev e | return v | . . .
Continuations K ::= init | K; λx. i
Machine States S ::= K  i | answer v
Here, init is the initial continuation, indicating that nothing further remains to
be done. The machine state answer v represents the final value of a computation
sequence. Based on the general consideration, we have the following transitions of
the abstract machine.
S =⇒ S 0

S goes to S 0 in one computation step

init  return v =⇒ answer v

st init

st return
K; λx. i  return v =⇒ K  [v/x]i
Further rules arise from considering each expression constructor in turn, possibly
adding new special-purpose intermediate instructions. We will write the rules in the
form label :: S =⇒ S 0 as a more concise alternative to the format used above. The
meaning, however, remains the same: each rules is an axiom defining the transition
judgment. First, the constructors:
 ev z
 ev (s e)

st z :: K
st s :: K

=⇒ K

=⇒ K; λx. return (s x) 

return z
e

Second, the corresponding destructor:
st case

 ev (case e1 of z ⇒ e2 | s x ⇒ e3 )
=⇒ K; λx1 . case1 x1 of z ⇒ e2 | s x ⇒ e3  ev e1
 case1 z of z ⇒ e2 | s x ⇒ e3
=⇒ K  ev e2
 case1 s v1 of z ⇒ e2 | s x ⇒ e3 =⇒ K  ev [v10 /x]e3

:: K

st case1 z :: K
st case1 s :: K

We can see that the case construct requires a new instruction of the form
case1 v1 of z ⇒ e2 | s x ⇒ e3 . This is distinct from case e1 of z ⇒ e2 | s x ⇒ e3
in that the case subject is known to be a value. Without an explicit new construct,
computation could get into an infinite loop since every value is also an expression
which evaluates to itself. It should now be clear how pairs and projections are
computed; the new instructions are hv1 , e2 i1 , fst1 , and snd1 .
st
st
st
st
st
st

pair
pair1
fst
fst1
snd
snd1

::
::
::
::
::
::

K
K
K
K
K
K








ev he1 , e2 i
hv1 , e2 i1
ev (fst e)
fst1 hv1 , v2 i
ev (snd e)
snd1 hv1 , v2 i

=⇒
=⇒
=⇒
=⇒
=⇒
=⇒

K; λx1 . hx1 , e2 i1
K; λx1 . return hv1 , x2 i
K; λx. fst1 x
K
K; λx. snd1 x
K








ev e1
ev e2
ev e
return v1
ev e
return v2

Neither functions, nor definitions or recursion introduce any essentially new
ideas. We add two new forms of instructions, app1 and app2 , for the intermediate
forms while evaluating applications.
K
K
K
K






ev (lam x. e)
ev (e1 e2 )
app1 v1 e2
app2 (lam x. e01 ) v2

=⇒
=⇒
=⇒
=⇒

st letv
st letn

:: K
:: K




let val x = e1 in e2
let name u = e1 in e2

=⇒ K; λx1 . [x1 /x]e2  ev e1
=⇒ K  ev ([e1 /x]e2 )

st fix

:: K



fix u. e

=⇒ K  ev ([fix u. e/u]e)

st
st
st
st

lam
app
app1
app2

::
::
::
::

K  return lamx. e
K; λx1 . app1 x1 e2  ev e1
K; λx2 . app2 v1 x2  ev e2
K  ev ([v2 /x]e01 )

The complete set of instructions as extracted from the transitions above:
Instructions i ::= ev e | return v
| case1 v1 of z ⇒ e2 | s x ⇒ e3
| hv1 , e2 i1 | fst1 v | snd1 v
| app1 v1 e2 | app2 v1 v2

Natural numbers
Pairs
Functions

The implementation of instructions, continuations, and machine states in Elf uses
infix operations to make continuations and states more readable.
% Machine Instructions
inst : type. %name inst I.
ev : exp -> inst.
return : exp -> inst.
case1 : exp -> exp -> (exp -> exp) -> inst.
pair1 : exp -> exp -> inst.
fst1 : exp -> inst.
snd1 : exp -> inst.
app1 : exp -> exp -> inst.
app2 : exp -> exp -> inst.
% Continuations
cont : type. %name cont K.
init : cont.
;
: cont -> (exp -> inst) -> cont.
%infix left 8 ;.
% Continuation Machine States
state : type. %name state S.
# : cont -> inst -> state.
answer : exp -> state.
%infix none 7 #.
The following declarations constitute a direct translation of the transition rules
above.
=> : state -> state -> type.
%infix none 6 =>.
%mode => +S -S’.

%name => C.

% Natural Numbers
st_z : K # (ev z) => K # (return z).
st_s : K # (ev (s E’)) => (K ; [x’] return (s x’)) # (ev E’).
st_case : K # (ev (case E1 E2 E3)) => (K ; [x1] case1 x1 E2 E3) # (ev E1).
st_case1_z : K # (case1 (z) E2 E3) => K # (ev E2).
st_case1_s : K # (case1 (s V1’) E2 E3) => K # (ev (E3 V1’)).
% Pairs
st_pair : K # (ev (pair E1 E2)) => (K ; [x1] pair1 x1 E2) # (ev E1).
st_pair1 : K # (pair1 V1 E2) => (K ; [x2] return (pair V1 x2)) # (ev E2).
st_fst : K # (ev (fst E’)) => (K ; [x’] fst1 x’) # (ev E’).
st_fst1 : K # (fst1 (pair V1 V2)) => K # (return V1).
st_snd : K # (ev (snd E’)) => (K ; [x’] snd1 x’) # (ev E’).
st_snd1 : K # (snd1 (pair V1 V2)) => K # (return V2).
% Functions
st_lam : K # (ev (lam E’)) =>
st_app : K # (ev (app E1 E2))
st_app1 : K # (app1 V1 E2) =>
st_app2 : K # (app2 (lam E1’)

K # (return (lam E’)).
=> (K ; [x1] app1 x1 E2) # (ev E1).
(K ; [x2] app2 V1 x2) # (ev E2).
V2) => K # (ev (E1’ V2)).

% Definitions
st_letv : K # (ev (letv E1 E2)) => (K ; [x1] ev (E2 x1)) # (ev E1).
st_letn : K # (ev (letn E1 E2)) => K # (ev (E2 E1)).
% Recursion
st_fix : K # (ev (fix E’)) => K # (ev (E’ (fix E’))).
% Return Instructions
st_return : (K ; [x] I x) # (return V) => K # (I V).
st_init : (init) # (return V) => (answer V).

Multi-step computation sequences could be represented as lists of single step
transitions. However, we would like to use dependent types to guarantee that, in
a valid computation sequence, the result state of one transition matches the start
state of the next transition. This is difficult to accomplish using a generic type of
lists; instead we introduce specific instances of this type which are structurally just
like lists, but have strong internal validity conditions.

∗

S =⇒ S 0
c
e ,→ v

S goes to S 0 in zero or more steps
e evaluates to v using the continuation machine

∗

∗

S =⇒ S 0

stop

S 0 =⇒ S 00
∗

S =⇒ S 00

S =⇒ S

step

∗

init  ev e =⇒ answer v
c

cev

e ,→ v
We would like the implementation to be operational, that is, queries of the
form ?- ceval peq V. should compute the value V of a given e. This means the
S =⇒ S 0 should be the first subgoal and hence the second argument of the step
rule. In addition, we employ a visual trick to display computation sequences in a
readable format by representing the step rule as a left associative infix operator.
=>* : state -> state -> type.
%infix none 5 =>*.
%mode =>* +S -S’.

%name =>* C*.

stop : S =>* S.
<< : S =>* S’’
<- S => S’
<- S’ =>* S’’.
%infix left 5 <<.
% Because of evaluation order, computation sequences are displayed
% in reverse, using "<<" as a left-associative infix operator.
ceval : exp -> exp -> type.
%mode ceval +E -V.

%name ceval CE.

cev : ceval E V
<- (init) # (ev E) =>* (answer V).
We then get a reasonable display of the sequence of computation steps which
must be read from right to left.
?- C* : init # (ev (app (lam [x] x) z)) =>* answer V.
Solving...
V = z.
C* =
stop << st_init << st_z << st_app2 << st_return << st_z << st_app1
<< st_return << st_lam << st_app.
c

The overall task now is to prove that e ,→ v if and only if e ,→ v. In one
direction we have to find a translation from tree-structured derivations D :: e ,→ v
∗
to sequential computations C :: init  ev e =⇒ answer v. In the other direction

we have to find a way to chop a sequential computation into pieces which can be
reassembled into a tree-structured derivation.
We start with the easier of the two proofs. We assume that e ,→ v and try to show
c
∗
that e ,→ v. This immediately reduces to showing that init  ev e =⇒ answer v.
This does not follow directly by induction, since subcomputations will neither start
from the initial computation nor return the final answer. If we generalize the claim
∗
to state that for all continuations K we have that K  ev e =⇒ K  return v,
then it follows directly by induction, using some simple lemmas regarding the concatenation of computation sequences (see Exercise 6.17).
We can avoid explicit concatenation of computation sequences and obtain a more
direct proof (and more efficient program) if we introduce an accumulator argument.
This argument contains the remainder of the computation, starting from the state
K  return v. To the front of this given computation we add the computation
∗
from K  ev e =⇒ K  return v, passing the resulting computation as the next
value of the accumulator argument. Translating this intuition to a logical statement
requires explicitly universally quantifying over the accumulator argument.
Lemma 6.22 For any closed expression e, value v and derivation D :: e ,→ v, if
∗
∗
C 0 :: K  return v =⇒ answer w for any K and w, then C :: K  ev e =⇒
answer w.
Proof: The proof proceeds by induction on the structure of D. Since the accumulator argument must already hold the remainder of the overall computation upon
appeal to the induction hypothesis, we apply the induction hypothesis on the immediate subderivations of D in right-to-left order.
The proof is implemented by a type family
ccp : eval E V
-> K # (return V) =>* (answer W)
-> K # (ev E) =>* (answer W)
-> type.
%mode ccp +D +C’ -C.
Operationally, the first argument is the induction argument, the second argument
the accumlator, and the last the output argument.
We only show a couple of cases in the proof; the others follow in a similar
manner.
Case:

D = lam x. e ,→ lam x. e ev lam
1
1
∗

C 0 :: K  return lam x. e1 =⇒ answer w
C :: K  ev(lam x. e1 ) =⇒ answer w

Assumption
By st lam followed by C 0

In this case we have added a step st lam to a computation; in the implemen∗
tation, this will be an application of the step rule for the S =⇒ S 0 judgment,
which is written as << in infix notation. Recall that the reversal of the evaluation order means that computations (visually) proceed from right to left.
ccp_lam : ccp (ev_lam) C’ (C’ << st_lam).

Case:
D=

D1
e1 ,→ lam x. e01

D2
e2 ,→ v2

D3
[v2 /x]e01 ,→ v

ev app

e1 e2 ,→ v
∗

C 0 :: K  return v =⇒ answer w
Assumption
∗
C3 :: K  ev ([v2 /x]e01 ) =⇒ answer w
By ind. hyp. on D3 and C 0
∗
C20 :: K; λx2 . app2 (lam x. e01 ) x2  return v2 =⇒ answer w
By st return and st app2 followed by C3
∗
0
C2 :: K; λx2 . app2 (lam x. e1 ) x2  ev e2 =⇒ answer w
By ind. hyp. on D2 and C20
∗
0
0
C1 :: K; λx1 . app1 x1 e2  return lam x. e1 =⇒ answer w
By st return and st app1 followed by C2 .
∗
C1 :: K; λx1 . app1 x1 e2  ev e1 =⇒ answer w
By ind. hyp. on D1 and C10
∗
C :: K  ev (e1 e2 ) =⇒ answer w
By st app followed by C1 .
The implementation threads the accumulator argument, adding steps concerned with application as in the proof above.
ccp_app : ccp
<<<-

(ev_app D3 D2 D1) C’ (C1 << st_app)
ccp D3 C’ C3
ccp D2 (C3 << st_app2 << st_return) C2
ccp D1 (C2 << st_app1 << st_return) C1.

2
From this, the completeness of the abstract machine follows directly.
Theorem 6.23 (Completeness of the Continuation Machine) For any closed exc
pression e and value v, if e ,→ v then e ,→ v.
Proof: We use Lemma 6.22 with K = init, w = v, and C 0 the computation with
∗
st init as the only step, to conclude that there is a computation C :: init  ev e =⇒
c
answer v. Therefore, by rule cev, e ,→ v.
The implementation is straightforward, using ccp, the implementation of the
main lemma above.

cpm_complete : eval E V -> ceval E V -> type.
%mode cpm_complete +D -C.
cpmcp : cpm_complete D (cev C)
<- ccp D (stop << st_init) C.

2
Now we turn our attention to the soundness of the continuation machine: whenever it produces a value v then the natural semantics can also produce the value v
from the same expression. This is more difficult to prove than completeness. The
reason is that in the completeness proof, every subderivation of D :: e ,→ v can
inductively be translated to a sequence of computation steps, but not every sequence of computation steps corresponds to an evaluation. For example, the partial
computation
∗
K  ev (e1 e2 ) =⇒ K; λx1 . app1 x1 e2  ev e1
represents only a fragment of an evaluation. In order to translate a computation
sequence we must ensure that it is sufficiently long. A simple way to accomplish this
is to require that the given computation goes all the way to a final answer. Thus, we
have a state K  ev e at the beginning of a computation sequence C to a final answer
w, there must be some initial segment of C 0 which corresponds to an evaluation of e
to a value v, while the remaining computation goes from K  return v to the final
answer w. This can then be proved by induction.
Lemma 6.24 For any continuation K, closed expression e and value w, if C ::
∗
K  ev e =⇒ answer w then there is a value v a derivation D :: e ,→ v, and a
∗
subcomputation C 0 of C of the form K  return v =⇒ answer w.
Proof: By complete induction on the structure of C. Here complete induction, as
opposed to a simple structural induction, means that we can apply the induction
hypothesis to any subderivation of C, not just to the immediate subderivations.
It should be intuitively clear that this is a valid induction principle (see also Section 6.4).
In the implementation we have chosen not to represent the evidence for the
assertion that C 0 is a subderivation of C. This can be added, either directly to the
implementation or as a higher-level judgment (see Exercise ??). This information
is not required to execute the proof on specific computation sequences, although it
is critical for seeing that it always terminates.
csd : K #
->
->
->
%mode csd

(ev E) =>* (answer W)
eval E V
K # (return V) =>* (answer W)
type.
+C -D -C’.

We only show a few typical cases; the others follow similarly.
∗

Case: The first step of C is st lam followed by C1 :: K  return lam x. e =⇒
answer w.
In this case we let D = ev lam and C 0 = C1 . The implementation (where step
is written as << in infix notation):
csd_lam : csd (C’ << st_lam) (ev_lam) C’.
∗

Case: The first step of C is st app followed by C1 :: K; λx1 . app1 x1 e2  ev e1 =⇒
answer w, where e = e1 e2 .
D1 :: e1 ,→ v1 for some v1 and
∗
C10 :: K; λx1 . app1 x1 e2  return v1 =⇒ answer w
By ind. hyp. on C1
∗
00
C1 :: K  app1 v1 e2 =⇒ answer w
By inversion on C10
∗
C2 :: K; λx2 . app2 v1 x2  ev e2 =⇒ answer w
By inversion on C100
D2 :: e2 ,→ v2 form some v2 and
∗
By ind. hyp. on C2
C20 :: K; λx2 . app2 v1 x2  return v2 =⇒ answer w
∗
00
C2 :: K  app2 v1 v2 =⇒ answer w
By inversion on C20
v1 = lam x. e01 and
∗
C3 :: K  ev ([v2 /x]e01 ) =⇒ answer w
By inversion on C200
0
D3 :: [v2 /x]e1 ,→ v for some v and
∗
C 0 :: K  return v =⇒ answer w
By ind. hyp on C3
D :: e1 e2 ,→ v
By rule ev app from D1 , D2 , and D3 .
The evaluation D and computation sequence C 0 now satisfy the requirements
of the lemma. The appeals to the induction hypothesis are all legal, since
C > C1 > C100 > C2 > C20 > C200 > C3 > C 0 , where > is the subcomputation
judgment. Each of the subcomputation judgments in this chain follows either
immediately, or by induction hypothesis.
The implementation:
csd_app : csd
<<<-

(C1
csd
csd
csd

<<
C1
C2
C3

st_app) (ev_app D3 D2 D1) C’
D1 (C2 << st_app1 << st_return)
D2 (C3 << st_app2 << st_return)
D3 C’.

2
Once again, the main theorem follows directly from the lemma.

Theorem 6.25 (Soundness of the Continuation Machine) For any closed expresc
sion e and value v, if e ,→ v then e ,→ v.
∗

Proof: By inversion, C :: init  ev e =⇒ answer v. By Lemma 6.24 there is a
∗
derivation D :: e ,→ v0 and C 0 :: init  return v0 =⇒ answer v for some v0 . By
0
0
inversion on C we see that v = v and therefore D satisfies the requirements of the
theorem.
cpm_sound : ceval E V -> eval E V -> type.
%mode cpm_sound +C -D.
cpmsd : cpm_sound (cev C) D
<- csd C D (stop << st_init).
%terminates {} (cpm_sound C D).

2
6.6

Type Preservation and Progress

So far we have concentrated on operational aspects of the translation from a big-step
to a small-step semantics; now we turn to issues of typing. The first property is type
preservation—a reprise of the same property for the big-step semantics. But the
reformulation of the semantics also allows us to express new language properties,
still at a high level of abstraction. One of the most important ones is progress. It
states that in any valid abstract machine state we can have one of two situations:
either we have already computed the final answer of the program, or we can make
progress by taking a further step.
Type preservation and progress together express that well-typed programs cannot go wrong, a phrase coined by Milner [Mil78]. In our setting, “going wrong”
corresponds to reaching a machine state in which no further transition rules applicable. Note that type preservation for the big-step semantics does not express this,
since it only talks about completed evaluations, not about intermediate states. Of
course, even in the small-step semantics an expression can fail to have a value, but
from the progress theorem we know that this is only due to non-termination.
We begin by giving the typing rules for the continuation-passing machine. One
complication as compared to the typing rules for expressions is that certain machine
states only make sense when components of instructions are values. For example,
the instruction app2 v1 v2 requires both arguments to be values. If this restriction
is not enforced, the progress theorem clearly fails, because there is no transition for
a state K  app2 ((lam x. x) (lam y. y)) z even though the arguments to app2
are correctly typed. This introduces a further complication: since continuations

are composed of functions from values to instructions, we need to record that the
argument of the continuation is indeed a function. For this we introduce a new
context Υ, which is either empty or contains a simple hypothesis x Value. The
value judgment is appropriately generalized so that x Value . x Value.
Υ ::= · | x Value Continuation argument
Υ; ∆ . i : τ
.K :τ ⇒σ
.S:σ

Instruction i has type τ in context ∆.
Continuation K maps values of type τ to answers of type σ
State S has type σ

Typing for continuations keeps track of two types: the type of the value that
will be passed to it, and the type of the final answer it produces. States on the
other hand record only the type of final answer it may produce. The judgments for
continuations and state do not depend on a context because they never contain free
variables.
∆.e:τ
Υ; ∆ . ev e : τ
∆ . v1 : nat

∆.v:τ

vi ev

Υ . v Value

Υ; ∆ . return v : τ

∆ . e2 : τ

∆, x:nat . e2 : τ

vi return

Υ . v1 Value

Υ; ∆ . (case1 v1 of z ⇒ e2 | s x ⇒ e3 ) : τ
∆ . v1 : τ1

∆ . e2 : τ2

Υ . v1 Value

Υ; ∆ . hv1 , e2 i1 : τ1 × τ2
∆ . v0 : τ1 × τ2

Υ . v0 Value

Υ; ∆ . fst1 v0 : τ1
∆ . v1 : τ2 → τ1

vi fst1

Υ . v0 Value

Υ; ∆ . snd1 v0 : τ2

∆ . e2 : τ2

∆ . v2 : τ2

vi pair1

∆ . v0 : τ1 × τ2

Υ . v1 Value

Υ; ∆ . app1 v1 e2 : τ1
∆ . v1 : τ2 → τ1

vi case1

Υ . v1 Value

Υ; ∆ . app2 v1 v2 : τ1

vi snd1

vi app1

Υ . v2 Value

vi app2

In the typing rules for continuations, we have to make sure that the parts of the
continuation are composed properly: the value returned by the last instruction of a
continuation matches the type accepted by the remaining continuation. The initial
continuation just returns its argument as the final answer and therefore has type
τ ⇒ τ.

. init : τ ⇒ τ
x Value; ·, x:τ . i : τ 0

vk init
. K : τ0 ⇒ σ

. K; λx. i : τ ⇒ σ

vk ;

For a state we verify that the type of the instruction to be executed matches
the one expected by the continuation, and assign the type of the final answer to the
state.
·; · . i : τ

.K :τ ⇒σ

. (K  i) : σ
·.v:σ

· . v Value

vs 

vs answer

. answer v : σ
The typing judgments for instructions, continuations, and states admit more
states as valid than can be reached from an initial state of the form init  ev e (see
Exercise 6.19). However, they are accurate enough to permit proof of preservation
and progress and is therefore appropriate for our purposes.
The implementation of the judgments in the logical framework is straightforward. The fact that we need at most one value variable is not explicitly represented.
Instead, we use the usual techniques for parametric and hypothetical judgments by
assuming value x for a new parameter x. The declarations below can be executed
in Elf in order to check the validity of instructions, continuations and states and
infer their most general types.
%%% Instructions
valid : inst -> tp -> type.
%mode valid +I *T.
% Evaluation and return
vi_ev : valid (ev E) T
<- of E T.
vi_return : valid (return V) T
<- of V T
<- value V.
% Natural Numbers
vi_case1 : valid (case1 V1 E2 E3) T
<- of V1 nat
<- of E2 T

%name valid VL.

<- ({x:exp} of x nat -> of (E3 x) T)
<- value V1.

% Pairs
vi_pair1 : valid (pair1 V1 E2) (cross T1 T2)
<- of V1 T1
<- of E2 T2
<- value V1.
vi_fst1 : valid (fst1 V’) T1
<- of V’ (cross T1 T2)
<- value V’.
vi_snd1 : valid (snd1 V’) T2
<- of V’ (cross T1 T2)
<- value V’.
% Functions
vi_app1 : valid (app1 V1 E2) T1
<- of V1 (arrow T2 T1)
<- of E2 T2
<- value V1.
vi_app2 : valid (app2 V1 V2) T1
<- of V1 (arrow T2 T1)
<- of V2 T2
<- value V1
<- value V2.
%%% Continuations
validk : cont -> tp -> tp -> type.
%mode validk +K *T *S.

%name validk VK.

vk_init : validk (init) T T.
vk_; : validk (K ; [x] I x) T S
<- ({x:exp} value x -> of x T -> valid (I x) T’)
<- validk K T’ S.
%%% States
valids : state -> tp -> type.

%name valids VS.

vs_# : valids (K # I) S
<- valid I T
<- validk K T S.
vs_answer : valids (answer V) S
<- of V S
<- value V.

With this preparation, we can now prove preservation and progress. The proofs

are quite straightforward, but somewhat tedious.
Theorem 6.26 (One-Step Type Preservation) If . S : σ and S =⇒ S 0 then . S 0 :
σ.
Proof: By cases on the derivation of C :: S =⇒ S 0 , applying several levels of
inversion to the given typing derivation for S. The implementation is via a type
family
vps : valids S T -> S => S’ -> valids S’ T -> type.
%mode vps +VS +C -VS’.
that relates the three derivation involved in the theorem. We show only a few
representative cases; the full implementation can be found in the on-line course
material.
Case: C is st app:
K  ev (e1 e2 ) =⇒ K; λx1 . app1 x1 e2  ev e1 .
. K  ev (e1 e2 ) : σ
. K : τ ⇒ σ and
·; · . ev (e1 e2 ) : τ for some τ
· . e1 e2 : τ
· . e1 : τ2 → τ and
· . e2 : τ2 for some τ2
x1 Value; x1 : τ2 → τ . app1 x1 e2 : τ
. (K; λx1 . app1 x1 e2 ) : (τ2 → τ ) ⇒ σ
. (K; λx1 . app1 x1 e2 )  ev e1 : σ

Assumption
By inversion
By inversion
By inversion
By rule (vi app1 )
By rule (vk ;)
By rule (vs )

The inversion steps are represented by expanding the structure of the first
argument, obtaining access to the subderivations VK of . K : τ → σ, P2 of
· . e2 : τ2 , and P1 of · . e1 : τ2 → τ . The needed derivation for S 0 is readily
constructed from these variables.
vps_app : vps (vs_# VK (vi_ev (tp_app P2 P1))) (st_app)
(vs_# (vk_; VK ([x1] [q1:value x1]
[p1:of x1 (arrow T2 T1)]
vi_app1 q1 P2 p1))
(vi_ev P1)).
Case: C is st app1 :
K  app1 v1 e2 =⇒ K; λx2 . app2 v1 x2  ev e2 .

. K  app1 v1 e2 : σ
. K : τ ⇒ σ and
·; · . app1 v1 e2 : τ for some τ
· . v1 : τ2 → τ and
· . e2 : τ2 form some τ2 and
· . v1 Value
x2 Value; x2 : τ2 .. app2 v1 x2 : τ
. (K; λx2 . app1 v1 e2 ) : τ2 ⇒ σ
. (K; λx1 . app2 v1 x2 )  ev e2 : σ

Assumption
By inversion

By inversion
By rule (vi app2 )
By rule (vk ;)
By rule vs cpm

In the representation we decompose the first argument as before. This now
gives as also Q1 which is the derivation of v1 Value.
vps_app1 : vps (vs_# VK (vi_app1 Q1 P2 P1)) (st_app1)
(vs_# (vk_; VK ([x2] [q2:value x2] [p2:of x2 T2]
vi_app2 q2 Q1 p2 P1))
(vi_ev P2)).
Case: C is st app2 :
K  app2 (lam x. e01 ) v2 =⇒ K  ev ([v2 /x]e01 ).
. (K  app2 (lam x. e01 ) v2 ) : σ
. K : τ1 ⇒ σ and
·; · . app2 (lam x. e01 ) v2 ) : τ1 for some τ1
· . (lam x. e01 ) : τ2 → τ1 and
· . v2 : τ2 for some τ2 and
· . (lam x. e01 ) Value and
· . v2 Value
x:τ2 . e01 : τ1
· . [v2 /x]e01 : τ1
·; · . ev ([v2 /x]e01 ) : τ1
. K  ev ([v2 /x]e01 ) : σ

Assumption
By inversion

By inversion
By inversion
By substitution property (2.4)
By rule (vi ev)
By rule (vs cpm)

By applying inversion as above we obtained
P10 :: (x:τ2 . e01 : τ1 )
which is represented by the variable
P1’ : {x:exp} of x T2 -> of (E1’ x) T1.

6.6. TYPE PRESERVATION AND PROGRESS

199

The appeal to the substitution principle is implemented by applying this function to the representation of v2 and the typing derivation P2 :: (· . v2 : τ2 ).
Note that we do not need the derivation Q2 which is evidence that v2 is a
value.
vps_app2 : vps (vs_# VK (vi_app2 Q2 (val_lam) P2 (tp_lam P1’)))
(st_app2) (vs_# VK (vi_ev (P1’ V2 P2))).

2
The multi-step type preservation theorem is a direct consequence of the one-step
preservation.
∗

Theorem 6.27 (Multi-Step Type Preservation) If . S : σ and S =⇒ S 0 then
. S0 : σ
∗

Proof: By straightforward induction on the structure of the derivation C ∗ :: (S =⇒
S 0 ). We only show the implementation.
vps* : valids S T -> S =>* S’ -> valids S’ T -> type.
%mode vps* +VS +C* -VS’.
vps*_stop : vps* VS (stop) VS.
vps*_<< : vps* VS (C2* << C1) VS2
<- vps VS C1 VS1
<- vps* VS1 C2* VS2.

2
Finally, we come to the progress theorem: we can make a transition from every
state that is not a final state. Recall that the only final states are of the form
answer v.
Theorem 6.28 (Progress) If . (K  i) : σ then there is a state S 0 such that
K  i =⇒ S 0 .
Proof: We know by inversion that
. K : τ ⇒ σ and
. i : τ for some τ .
We apply case analysis on i. In each case we can either directly make a transition,
or we need to apply several inversions on an available typing or value derivation
until each subcase can be seen to be impossible or a transition rule applies. For a
return instruction, we also need to distinguish cases on the shape of K. We show
only a few cases.

Case: i = ev (e1 e2 ). Then st app applies.
Case: i = app1 v1 e2 . Then st app1 applies.
Case: i = app2 v1 v2 . Then
· . v1
· . v2
· . v1
· . v2

Value and
Value and
: τ2 → τ and
: τ2

By inversion

Now we distinguish subcases on · . v2 Value
Subcase: v1 = z. This is impossible, since there is no rule to conclude
· . z : τ2 → τ .
Subcase: v1 = s v10 . This is impossible, since there is no rule to conclude
· . s v10 : τ2 → τ .
Subcase: v1 = hv10 , v100 i. This is impossible, since there is not rule to conclude
· . hv10 , v100 i : τ2 → τ
Subcase: v1 = lamx. e01 . Then st app2 applies.
The implementation is very similar to progress, except that we don’t need to construct the resulting typing derivation. Note that impossible cases are not represented. Also, for simplicity of implementation, we distinguish the cases on the
typing derivation rather than the instruction, which is possible since the typing
judgment is syntax-directed. We show only the three cases from above.
pgs : valids S T -> S => S’ -> type.
%mode pgs +VS -C.
pgs_app : pgs (vs_# VK (vi_ev (tp_app P2 P1))) (st_app).
pgs_app1 : pgs (vs_# VK (vi_app1 Q1 P2 P1)) (st_app1).
pgs_app2 : pgs (vs_# VK (vi_app2 Q2 (val_lam) P2 (tp_lam P1’)))
(st_app2).
Note that some applications of inversion may be redundant for the sake of uniformity. For example, we could have replaced the last clause by
pgs_app2’ : pgs (vs_# VK (vi_app2 Q2 (val_lam) P2 P1))
(st_app2).

Nonetheless, the inversion on the value derivation is necessary, and
pgs_app2’’ : pgs (vs_# VK (vi_app2 Q2 Q1 P2 P1))
(st_app2).
would be incorrect as a proof case because it is not apparent from the first argument
that st app2 indeed applies.
2

6.7

Contextual Semantics
[ This section discusses a contextual semantics as an alternative smallstep machine to the CPM machine. This still has to be revised from an
older version. ]

6.8

Exercises

Exercise 6.1 If we replace the rule ev app in the natural semantics of Mini-ML
(see Section 2.3) by
u
x ,→ v2
..
.
e1 ,→ lam x. e01

e2 ,→ v2
e1 e2 ,→ v

e01 ,→ v

ev app0

x,u

in order to avoid explicitly substituting v2 for x, something goes wrong. What
is it? Can you suggest a way to fix the problem which still employs hypothetical
judgments?
(Note: We assume that the third premiss of the modified rule is parametric in x
and hypothetical in u which is discharged as indicated. This implies that we assume
that x is not already free in any other hypothesis and that all labels for hypotheses
are distinct—so this is not the problem you are asked to detect.)
Exercise 6.2 Define the judgment W RealVal which restricts closures W to Λabstractions. Prove that · ` F ,→ W then W RealVal and represent this proof in
Elf.
Exercise 6.3 In this exercise we try to eliminate some of the non-determinism in
compilation.
1. Define a judgment F std which should be derivable if the de Bruijn expression
F is in the standard form in which the ↑ operator is not applied to applications
or abstractions.

2. Rewrite the translation from ordinary expressions e such that only standard
forms can be related to any expression e.
3. Prove the property in item 2.
4. Implement the judgments in items 1, 2, and the proof in item 3.
Exercise 6.4 Restrict yourself to the fragment of the language with variables, abstraction, and application, that is,
F

::= 1 | F ↑ | ΛF | F1 F2

1. Define a judgment F Closed that is derivable iff the de Bruijn expression F is
closed, that is, has no free variables at the object level.
2. Define a judgment for conversion of de Bruijn expressions F to standard form
(as in Exercise 6.3, item 1) in a way that preserves meaning (as given by its
interpretation as an ordinary expression e).
3. Prove that, under appropriate assumptions, this conversion results in a de
Bruijn expression in standard form equivalent to the original expression.
4. Implement the judgments and correctness proofs in Elf.
Exercise 6.5 Restrict yourself to the same fragment as in Exercise 6.4 and define
the operation of substitution as a judgment subst F1 F2 F . It should be a consequence of your definition that if ΛF1 represents lam x. e1 , F2 represents e2 , and
subst F1 F2 F is derivable then F should represent [e2 /x]e1 . Furthermore, such an
F should always exist if F1 and F2 are as indicated. With appropriate assumptions
about free variables or indices (see Exercise 6.4) prove these properties, thereby
establishing the correctness of your implementation of substitution.
Exercise 6.6 Write out the informal proof of Theorem 6.7.
Exercise 6.7 Prove Theorem 6.8 by appropriately generalizing Lemma 6.2.
Exercise 6.8 Standard ML [MTH90] and many other formulations do not contain a let name construct. Disregarding problems of polymorphic typing for the
moment, it is quite simple to simulate let name with let val operationally using
so-called thunks. The idea is that we can prohibit the evaluation of an arbitrary
expression by wrapping it in a vacuous lam-abstraction. Evaluation can be forced
by applying the function to some irrelevant value (we write z, most presentations
use a unit element). That is, instead of
l

= let name x = e1 in e2

l0

= let val x0 = lam y. e1 in [x0 z/x]e2

where y is a new variable not free in e1 .
1. Show a counterexample to the conjecture “If l is closed, l ,→ v, and l0 ,→ v0
then v = v0 (modulo renaming of bound variables)”.
2. Show a counterexample to the conjecture “. l : τ iff . l0 : τ ”.
3. Define an appropriate congruence e ∼
= e0 such that l ∼
= l0 and if e ∼
= e0 , e ,→ v
0
0
0
∼
and e ,→ v then v = v .
4. Prove the properties in item 3.
5. Prove that if the values v and v0 are natural numbers, then v ∼
= v0 iff v = v0 .
We need a property such as the last one to make sure that the congruence we define
does not identify all expressions. It is a special case of a so-called observational
equivalence (see ??).
Exercise 6.9 The rules for evaluation in Section 6.2 have the drawback that looking up a variable in an environment and evaluation are mutually recursive, since
the environment contains unevaluated expressions. Such expressions may be added
to the environment during evaluation of a let name or fix construct. In the definition of Standard ML [MTH90] this problem is avoided by disallowing let name
(see Exercise 6.8) and by syntactically restricting occurrences of the fix construct.
When translated into our setting, this restriction states that all occurrences of fixpoint expressions must be of the form fix x. lam y. e. Then we can dispense with
the environment constructor + and instead introduce a constructor ∗ that builds a
recursive environment. More precisely, we have
Environments η

::= · | η, W | η ∗ F

The evaluation rules fev 1+, fev ↑+, and fev fix on page 161 are replaced by
K ` fix0 F ,→ {K ∗ F ; F }
K ∗ F ` 1 ,→ {K ∗ F ; F }
K ` F ,→ W
K ∗ F 0 ` F ↑ ,→ W

fev fix∗

fev 1∗

fev ↑∗

1. Implement this modified evaluation judgment in Elf.
2. Prove that under the restriction that all occurrences of fix0 in de Bruijn expressions have the form fix0 ΛF for some F , the two sets of rules define an
equivalent operational semantics. Take care to give a precise definition of the
notion of equivalence you are considering and explain why it is appropriate.
3. Represent the equivalence proof in Elf.
4. Exhibit a counterexample which shows that some restriction on fixpoint expressions (as, for example, the one given above) is necessary in order to preserve equivalence.
5. Under the syntactic restriction from above we can also formulate a semantics
which requires no new constructor for environments by forming closures over
fixpoint expressions. Then we need to add another rule for application of
an expression which evaluates to a closure over a fixpoint expression. Write
out the rules and prove its equivalence to either the system above or the
original evaluation judgment for de Bruijn expressions (under the appropriate
restriction).
Exercise 6.10 Show how the effect of the bind instruction can be simulated in the
CLS machine using the other instructions. Sketch the correctness proof for this
simulation.
Exercise 6.11 Complete the presentation of the CLS machine by adding recursion.
In particular
1. Complete the computation rules on page 173.
2. Add appropriate cases to the proofs of Lemmas 6.16, and 6.18.
Exercise 6.12 Prove the following carefully.
1. The concatenation operation “◦” on computations is associative.
2. The subcomputation relation “<” is transitive (Lemma 6.17).
Show the implementation of your proofs as type families in Elf.
Exercise 6.13 The machine instructions from Section 6.3 can simply quote expressions in de Bruijn form and consider them as instructions. As a next step in
the (abstract) compilation process, we can convert the expressions to lower-level
code which simulates the effect of instructions on the environment and value stacks
in smaller steps.
1. Design an appropriate language of operations.

2. Specify and implement a compiler from expressions to code.
3. Prove the correctness of this step of compilation.
4. Implement your correctness proof in Elf.
Exercise 6.14 Types play an important role in compilation, which is not reflected
in the some of the development of this chapter. Ideally, we would like to take
advantage of type information as much as possible in order to produce more compact
and more efficient code. This is most easily achieved if the type information is
embedded directly in expressions (see Section ??), but at the very least, we would
expect that types can be assigned to intermediate expressions in the compiler.
1. Define typing judgments for de Bruijn expressions, environments, and values
for the language of Section 6.2. You may assume that values are always closed.
2. Prove type preservation for your typing judgment and the operational semantics for de Bruijn expressions.
3. Prove type preservation under compilation, that is, well-typed Mini-ML expressions are mapped to well-typed de Bruijn expressions under the translation
of Section 6.2.
4. What is the converse of type preservation under compilation. Does your typing
judgment satisfy it?
5. Implement the judgments above in Elf.
6. Implement the proofs above in Elf.
Exercise 6.15 As in Exercise 6.14:
1. Define a typing judgment for evaluation contexts. It should only hold for valid
evaluation contexts.
2. Prove that splitting a well-typed expression which is not a value always succeeds and produces a unique context and redex.
3. Prove that splitting a well-typed expression results in a valid evaluation context and valid redex.
4. Prove the correctness of contextual evaluation with respect to the natural
semantics for Mini-ML.
5. Implement the judgments above in Elf. Evaluation contexts should be represented as functions from expressions to expressions satisfying an additional
judgment.

6. Implement the proofs above in Elf.
Exercise 6.16 Show that the purely expression-based natural semantics of Section 2.3 is equivalent to the one based on a separation between expressions and
values in Section 6.5. Implement your proof, including all necessary lemmas, in Elf.
Exercise 6.17 Carry out the alternative proof of completeness of the continuation
machine sketched on page 189. Implement the proof and all necessary lemmas in
Elf.
Exercise 6.18 Do the equivalence proof in Lemma 6.22 and the alternative in Exercise 6.17 define the same relation between derivations? If so, exhibit the bijection
in the form of a higher-level judgment relating the Elf implementations. Be careful
to write out necessary lemmas regarding concatenation. You may restrict yourself
to functional abstraction, application, and the necessary computation rules.
Exercise 6.19 Not every valid state of the CPM machine (according to the typing
judgments in Section 6.6) can be reached by a computation starting from some
initial state of the form init  ev e where · ` e : τ .
1. Exhibit a valid, but unreachable state.
2. Modify the validity judgments so that every valid machine state can in fact
be reached from some initial state.
3. Prove this property.
4. Implement your proof in Elf.

